

Werken met de MSX

praktijk- programma's

Robert-Jan Donkers

TERMINAL SOFTWARE PUBLICATIES

WERKEN MET MSX COMPUTERS
PRAKTIJKPROGRAMMA'S

WERKEN MET MSX COMPUTERS
PRAKTIJKPROGRAMMA'S

Robert-Jan Donkers

TERMINAL SOFTWARE
PUBLICATIES

Een uitgave van: Terminal Software Publicaties
Postbus 111,5110 AC Baarle Nassau

1e druk oktober 1985

(c) 1985, R.-J. Donkers

ISBN 90-6883-009-0

CIP-GEGEVENS KONINKLIJKE BIBLIOTHEEK, DEN HAAG

Donkers, Robert-Jan

Werken met de MSX computer: paktijkprogramma's / Robert-Jan
Donkers.- Baarle Nassau : Terminal Software Publicaties
ISBN 90-6883-009-0
SISO 365.3 UDC 681.3.06 UGI 650
Trefw.: computerprogramma's / MSX (computer).

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar
gemaakt worden, anders dan voor eigen gebruik, door middel van
druk, fotokopie, microfilm, elektronische informatiedragers of
op welke andere wijze dan ook, zonder voorafgaande schriftelijke
toestemming van de uitgever.

Ondanks alle zorg waarmee deze uitgave is samengesteld kan noch
de auteur noch de uitgever enige aansprakelijkheid aanvaarden
voor eventuele schade die zou kunnen voortvloeien uit het ge-
bruik van deze programma's of uit enige fout die in deze uitga-
ve zou kunnen voorkomen.

MSX is een handelsmerk van MICROSOFT CORP.

INHOUD

Voorwoord	7
Huishoudboekje	9
Autokosten	19
Music-O-Theek	30
Ledenadministratie	40
Omrekenen Angelsaksisch naar Metriek Stelsel en omgekeerd	48
Weerstandkleurcodes	58
Elementenlijst	62
Hi-Res Tekenen	70
Grafiek Plotten	77
Literatuuronderzoek	84
Klanten-Informatie	99
Telefoonindex	110
Buro-Agenda	119

VOORWOORD

Met de komst van de MSX computers is er nu eindelijk een gestandaardiseerd systeem voor home-computers op de markt.

De standaardisatie is niet alleen belangrijk voor de onderlinge uitwisselbaarheid van de programma's, maar ook de hardware is onderling uitwisselbaar. Bovendien is het MSX Systeem zodanig opgebouwd dat het niet alleen geschikt is om allerlei spelprogramma's mee te spelen, maar is uitermate geschikt voor allerlei serieuze zaken. In de nabije toekomst zal het MSX Systeem een steeds belangrijker rol gaan spelen in het huis van de toekomst.

Met de collectie programma's in dit boek wil ik dan ook de serieuze gebruiker een aantal kant en klare programma's bieden, waarmee hij of zij direct aan de slag kan. De programma's zijn allemaal geschreven in MSX BASIC en werken met een schermbreedte van 40 tekens. Heeft U geen monitor tot Uw beschikking, dan kan het noodzakelijk zijn de breedte op 37 tekens te zetten, omdat bij een normale TV er aan de zijkanten een aantal tekens kunnen afvallen.

De programma's zijn zodanig geschreven, dat U de ingevoerde gegevens zowel op cassette als diskette kunt opslaan.

Er zijn in de programma's geen modules opgenomen om allerlei foutmeldingen op te vangen, die kunt U er desgewenst zelf aan toevoegen. Beslist nodig is het niet, want de programma's werken foutloos, zoals ze hier in dit boek zijn afgedrukt. Als U de instructies opvolgt kan er niets mis gaan. De MENU's spreken voor zich.

U zult zich de programma's snel eigen maken, mede door het feit dat U ze zelf intoetst. Aldus leert U ook nog het gebruik van allerlei routines, die U misschien in Uw eigen programma's kunt toepassen.

Ik hoop dan ook dat U veel plezier en vooral veel gemak van deze programma's zult hebben. Voor eventuele opmerkingen Uwerzijds houd ik me gaarne aanbevolen. Een briefje naar de uitgever is dan voldoende.

September 1985,

Robert-Jan Donkers.

HUISHOUDBOEKJE

Met dit programma bent U in staat Uw vaste uitgaven per maand/kwartaal/jaar bij te houden. Ook Uw inkomsten worden per maand in het maandoverzicht bijgehouden.

Direct na het opstarten van het programma kunt U Uw salaris, vakantiegeld en een eventuele dertiende maand invoeren.

Daarna komt U in het Hoofdmenu.

Bij het invoeren van de diverse kostenposten kunt U elke kostensoort zijn eigen naam geven, met dien verstande dat die naam maximaal 17 tekens lang mag zijn.

U kunt 50 verschillende kostensoorten invoeren.

Het invoeren van de diverse kosten:

Bij het invoeren moet U aangeven of deze post elke maand of 4* ,6* of 1* per jaar moet worden afgerekend.

Als U bijv. invoert 12x per jaar, dan betekent dat er elke maand een bepaald bedrag moet worden afgerekend, zoals huur of hypotheek. Direct erna geeft U aan wat dat bedrag per maand is.

Dan wordt er gevraagd of er ook nog een jaarafrekening komt, zoals dat bijv. geldt voor gas,electriciteit of water. Is er geen afrekening dan toetst U een nul in en is er wel een jaarafrekening dan geeft U het bedrag en de maand waarin deze afrekening plaatsvindt.

Nadat U alle kosten heeft ingevoerd kunt U via de keuze 'gegevens opslaan' de gegevens wegschrijven naar tape of disk.

Zijn er bepaalde kosten die niet meer van toepassing zijn, dan kunt U die wissen.

Het programma is zo gebruikersvriendelijk mogelijk gehouden, zodat bediening vlot verloopt.

Bovendien is het programma van duidelijke REM-statements voorzien, zodat U zich snel een inzicht over de opbouw van het programma kunt verschaffen.

Eenmaal ingetoetst en op disk of tape gezet zult U met veel plezier van het programma gebruik maken, want nu kunt U elke maand zien waar U aan toe bent.

Mogelijke posten die U kunt gebruiken:

- Huur of Hypotheek
- Gas
- Water
- Electriciteit
- Autoverzekering
- W.A. Verzekering
- Brandverzekering
- Telefoon
- Ziekenfonds
- Omroepbijdrage
- Dagblad
- Onroer.goed.bel.
- Weekblad of leesmap
- Abon. MSX Blad
- Hobbyclub Moeder
- etc.etc.


```

10 REM *****
    **          HUISHOUDBOEKJE          **
    ** (C)1985 ROBERT-JAN DONKERS**
    **  VERSIE V3.2 VOOR MSX          **
    *****
20 REM *****
    **          INITIALISEREN          **
    *****
30 CLEAR 6000
40 DATA JAN.,FEB.,MRT.,APR.,MEI.,JUN.,JU
L.,AUG.,SEP.,OKT.,NOV.,DEC.
50 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
60 DIM N$(51),M$(51,12),G$(51),J$(51),I$
(51),F(51),S$(12)
70 ST=0
80 FOR N=1 TO 51
90 N$(N)="* ....."
100 NEXT N:GOTO 190
110 LOCATE 0,10:INPUT"Uw Maandsalaris : "
;K$
120 LOCATE 0,12:INPUT"In welke maand (1/
12) ontvangt U Uw          VAKANTIEGELD :";U
130 LOCATE 0,14:INPUT"Hoeveel is dat :";
L$
140 LOCATE 0,16:INPUT"Ontvangt U ook nog
een 13e maand :          zo ja,geef dan bed
rag, anders 0 ";W$
150 IF VAL(W$)>0 THEN LOCATE 0,18:INPUT"
In welke maand :";W
160 FOR N=1 TO 12:S$(N)=K$:NEXT N
170 LET S$(U)=STR$(VAL(S$(U))+VAL(L$))
180 LET S$(W)=STR$(VAL(S$(W))+VAL(W$))
:FOR I=1 TO 12:ST=ST+VAL(S$(I)):NEXT I
190 REM *****
    **          HOOFDMENU          **
    *****
200 CLS

```

```

210 PRINT"                                %%%%%%%%%%
                                         %           %
                                         % M E N U %
                                         %           %
                                         %%%%%%%%%%"

220 PRINT:PRINT
230 PRINT:PRINT"      Maandoverzicht-----
----->= 1"
240 PRINT:PRINT"      Jaaroverzicht-----
----->= 2"
250 PRINT:PRINT"      Uitgaven invoeren---
----->= 3"
260 PRINT:PRINT"      Uitgaveposten afvoer
en---->= 4"
270 PRINT:PRINT"      Uitgaveposten opslaa
n----->= 5"
280 PRINT:PRINT"      Uitgaveposten inleze
n----->= 6"
290 PRINT:PRINT"      Stoppen met programm
a----->= 7"
300 LOCATE 0,22:PRINT"          TOETS UW K
EUZE IN (1/7)"
310 Q$=INKEY$:IF Q$="" THEN 310
320 IF Q$<"1" OR Q$>"7" THEN 310
330 ON VAL(Q$) GOTO 680,1240,340,960,140
0,1760,2050
340 REM *****
      ** UITGAVEPOSTEN INVOEREN **
      *****

350 IF S$(1)=" " THEN CLS:GOTO 110
360 CLS
370 PRINT"          UITGAVEPOSTEN INVOEREN":
PRINT
380 PRINT STRING$(39,"%"):PRINT
390 FOR N=1 TO 50:IF N$(N)="*.....
...." THEN 420
400 NEXT N
410 PRINT:PRINT"BESTAND IS VOL !!!!":FOR
  Q=1 TO 1500:NEXT Q:GOTO 200
420 INPUT"OMSCHRIJVING  :";Q$

```

```

430 IF Q$="" THEN LOCATE 0,5:GOTO 420
440 IF LEN(Q$)>16 THEN PRINT"OMSCHRIJVING
TE LANG (MAX.16) !":LOCATE 0,5:GOTO 420
450 LL=LEN(Q$)
460 MID$(N$(N),1,LL)=Q$
470 PRINT
480 INPUT"FREQUENTIE PER JAAR : ";F(N)
490 IF F(N)>12 THEN PRINT"REKEN BEDRAG OM
PER MAAND !!":GOTO 480
500 IF F(N)=12 THEN RESTORE:FOR I=1 TO 12:READ A$:M$(N,I)=A$:NEXT I:GOTO 570
510 PRINT:PRINT"IN DE MAAND(EN):"
520 FOR T=1 TO F(N)
530 INPUT"MAANDNR.(1 I/M 12):";Q
540 RESTORE:FOR I=1 TO Q:READ K$:NEXT I:
M$(N,T)=K$
550 PRINT M$(M,T)
560 NEXT T
570 PRINT:INPUT"TERMIJNBEDRAG :";G$(N)
580 PRINT:INPUT"JAARAFREKENING(0=GEEN):"
;J$(N)
590 IF VAL(J$(N))=0 THEN 630
600 INPUT"IN WELKE MAAND ? (1 t/m 12) ";
I
610 IF I<1 OR I>12 THEN 600
620 RESTORE:FOR F=1 TO I:READ A$:NEXT F:
I$(N)=A$
630 LOCATE 0,22:PRINT" GEGEVENS IN ORD
E ? (Ja/Nee) "
640 Q$=INKEY$:IF Q$="" THEN 640
650 IF Q$="J" OR Q$="j" THEN 190
660 IF Q$="N" OR Q$="n" THEN CLS:PRINT"
UITGAVEPOSTEN INVOEREN":PRINT:PRINT S
TRING$(39,"%"):PRINT:GOTO 420
670 GOTO 640
680 REM *****
* M A A N D O U E R Z I C H T *
*****
690 H=0:CLS

```

```

700 PRINT" UITGAVEN IN EEN BEPAALDE MAAN
D":PRINT
710 PRINT STRING$(39,"%"):PRINT
720 INPUT"WELKE MAAND (1 t/m 12) :";K
730 IF K<1 OR K>12 THEN 690
740 RESTORE:FOR N=1 TO K:READ K$:NEXT N
750 CLS:LOCATE 16,0:PRINT"MAAND:";K$:PRI
NT USING" SALARIS: #####.##";VAL(S$(K)):
PRINT
760 FOR N=1 TO 50
770 FOR I=1 TO 12
780 IF N$(N)="*....." THEN 820
790 IF M$(N,I)=K$ THEN PRINT N$(N);:PRIN
T USING"F1 #####.##";VAL(G$(N)):H=H+VAL(
G$(N)):GOTO 810
800 NEXT I
810 NEXT N
820 KOP=0
830 FOR N=1 TO 50
840 IF N$(N)="*....."THEN 890
850 IF I$(N)<>K$ THEN 880
860 IF KOP=0 THEN PRINT:PRINT"JAARAFREKE
NING:":LET KOP=1
870 PRINT N$(N);:PRINT USING"F1 #####.##
";VAL(J$(N)):H=H+VAL(J$(N))
880 NEXT N
890 PRINT:PRINT STRING$(39,"%"):PRINT
900 PRINT"TOTAAL          ";:PRINT USING
"F1#####.##";H
910 LOCATE 0,22:PRINT"      >M< =MENU      >A
< = ANDERE MAAND"
920 Q$=INKEY$:IF Q$="" THEN 920
930 IF Q$="M" OR Q$="m" THEN 190
940 IF Q$="A" OR Q$="a" THEN 680
950 GOTO 920
960 REM *****
      **   UITGAVEPOSTEN AFVOEREN   **
      *****
970 CLS

```



```

980 LOCATE 0,21:PRINT"                >> M <<
    = MENU"
990 LOCATE 0,3:INPUT"WELKE RUBRIEK WILT
U AFVOEREN ";K$
1000 IF K$="M" OR K$="m" THEN 190
1010 LL=LEN(K$):IF LL>16 THEN 970
1020 LOCATE 0,21:PRINT SPACES(38)
1030 FOR N=1 TO 50:IF K$=MID$(N$(N),1,LL
) THEN 1060
1040 NEXT N
1050 LOCATE 0,3:PRINT K$;" zit niet in h
et bestand !":FOR V=1 TO 1500:NEXT V:GOT
O 960
1060 LOCATE 0,3:PRINT"OMSCHRIJVING :";N$
(N)
1070 LOCATE 0,21:PRINT" TOETS IN .....>
W< =WISSEN   >M< =MENU"
1080 Q$=INKEY$:IF Q$="" THEN 1080
1090 IF Q$="M" OR Q$="m" THEN 190
1100 IF Q$="W" OR Q$="w" THEN 1120
1110 GOTO 1080
1120 REM *****
      ** R U B R I E K - W I S S E N **
      *****
1130 FOR I=1 TO 51:IF N$(T)="-*.....
....." THEN 1150
1140 NEXT T
1150 N$(N)=N$(T-1):N$(T-1)=N$(T):F(N)=F(
T-1):F(T-1)=F(T)
1160 FOR M=1 TO 12:M$(N,M)=M$(T-1,M):M$(
T-1,M)=M$(T,M):NEXT M
1170 G$(N)=G$(T-1):G$(T-1)=G$(T):J$(N)=J
$(T-1):J$(T-1)=J$(T)
1180 I$(N)=I$(T-1):I$(T-1)=I$(T)
1190 LOCATE 4,10:PRINT" DE GEGEVENS ZIJN
GEWIST !"
1200 Q$=INKEY$:IF Q$="" THEN 1200
1210 IF Q$="M" OR Q$="m" THEN 190
1220 IF Q$="W" OR Q$="w" THEN 960
1230 GOTO 1200

```

```

1240 REM *****
      ** J A A R O U E R Z I C H T **
      *****

1250 CLS
1260 PRINT" J A A R O U E R Z I C H T"
1270 PRINT:PRINT USING"JAARSALARIS: ####
##.##";ST
1280 TT=0
1290 FOR N=1 TO 51
1300 H=0
1310 IF N$(N)="*....." THEN 13
60
1320 H=H+(F(N)*VAL(G$(N)))+VAL(J$(N))
1330 PRINT N$(N);:PRINT USING"F1 #####.#
#";H
1340 TT=TT+H
1350 NEXT N
1360 PRINT:PRINT"JAARTOTAAL ";:PRIN
T USING"F1#####.##";TT
1370 LOCATE 0,22:PRINT " >>> M
<<< = MENU"
1380 QS=INKEY$: IF QS="" THEN 1380
1390 IF QS="M" OR QS="m" THEN 190 ELSE 1
380
1400 REM *****
      ** GEGEVENS OPSLAAN **
      *****

1410 FOR K=1 TO 50:IF N$(K)="*.....
....." THEN NF=K: GOTO 1430
1420 NEXT K
1430 CLS
1440 PRINT:PRINT"MET WELKE BAUDRATE WILT
U NAAR CASSETTE SCHRIJVEN ? (1=1200 , 2
=2400 BAUD"
1450 QS=INKEY$: IF QS="" THEN 1450
1460 IF QS="1" THEN BR=1:GOTO 1490
1470 IF QS="2" THEN BR=2:GOTO 1490
1480 GOTO 1450
1490 PRINT:PRINT"GEEF NAAM VAN DE FILE:(
MAX.6 LETTERS)

```

```

1500 SCREEN, , , BR
1510 INPUT Q$
1520 SS$="CAS:"+Q$
1530 PRINT"ZET RECORDER KLAAR (RECORD+PLAY)"
1540 PRINT"TOETS DAARNA 'RETURN' VOOR SAUEN"
1550 Q$=INKEY$: IF Q$="" THEN 1550
1560 IF ASC(Q$)=13 THEN 1570 ELSE 1550
1570 OPEN SS$ FOR OUTPUT AS#1
1580 MOTOR ON
1590 PRINT#1,NF
1600 FOR K=1 TO NF
1610 PRINT#1,F(K)
1620 PRINT#1,G$(K)
1630 PRINT#1,I$(K)
1640 PRINT#1,N$(K)
1650 NEXT K
1660 FOR K=1 TO NF:FOR M=1 TO 12
1670 PRINT#1,M$(K,M)
1680 NEXT M:NEXT K
1690 FOR K=1 TO 12
1700 PRINT#1,S$(K)
1710 NEXT K
1720 PRINT#1,ST
1730 CLOSE#1
1740 MOTOR OFF
1750 GOTO 190
1760 REM *****
      **   GEGEVENS   INLEZEN   **
      *****

1770 CLS
1780 PRINT:PRINT"GEEF NAAM VAN DE FILE:(MAX.6LETTERS)"
1790 INPUT Q$
1800 IF LEN(Q$)>6 THEN 1770
1810 SS$="CAS:"+Q$
1820 PRINT"ZET RECORDER KLAAR (PLAY)"
1830 PRINT"TOETS DAARNA 'RETURN' VOOR INLEZEN"

```

```

1840 Q$=INKEY$: IF Q$="" THEN 1840
1850 IF ASC(Q$)=13 THEN 1860 ELSE 1840
1860 OPEN SS$ FOR INPUT AS#1
1870 MOTOR ON
1880 INPUT#1,NF
1890 FOR K=1 TO NF
1900 INPUT#1,F(K)
1910 INPUT#1,G$(K)
1920 INPUT#1,I$(K)
1930 INPUT#1,N$(K)
1940 NEXT K
1950 FOR K=1 TO NF:FOR M=1 TO 12
1960 INPUT#1,M$(K,M)
1970 NEXT M:NEXT K
1980 FOR K=1 TO 12
1990 INPUT#1,S$(K)
2000 NEXT K
2010 INPUT#1,ST
2020 CLOSE#1
2030 MOTOR OFF
2040 GOTO 190
2050 REM *****
      ** EINDE VAN 'T PROGRAMMA **
      *****

2060 CLS
2070 PRINT"U WILT HET PROGRAMMA BEEINDIG
EN ?"
2080 PRINT:PRINT"HEEFT U DE GEGEVENS DIE
U HEEFT INGEVOERD AL OP CASSETT
E GEZET ?"
2090 PRINT:PRINT"ALS U DIT AL GEDAAN HEE
FT OF ALS DAT NIET NODIG IS; TOETS DA
N 'E'(EINDE)"
2100 PRINT:PRINT"IN ALLE ANDERE GEVALLEN
TOETST U 'M', ZODAT U TERUGKEERT NAAR
HET HOOFDMENU"
2110 LOCATE 0,22:PRINT"TOETS UW KEUZE IN
('E' OF 'M')"
2120 Q$=INKEY$: IF Q$="" THEN 2120
2130 IF Q$="M" OR Q$="m" THEN 190
2140 IF Q$="E" OR Q$="e" THEN END
2150 GOTO 2120

```


AUTOKOSTEN

Met dit programma houdt U per maand de vaste en variabele kosten van Uw auto bij. Met de waardevermindering van Uw auto wordt geen rekening gehouden.

In dit programma kunt U de volgende vijf kostensoorten invoeren t.w.:

- Belasting,
- Brandstof,
- Onderhoud,
- Verzekering,
- Reparaties.

Kosten voor lidmaatschap ANWB en WW kunt U het beste opnemen onder 'Verzekeringen'.

Aan het begin van het programma geeft U de kilometerstand van het moment dat U met dit programma gaat werken.

Nadat U gedurende een jaar de gegevens heeft vastgelegd, geeft U de eindstand.

Uw MSX berekent dan voor U het aantal gereden kilometers, het totaal der kosten en daaruit wordt dan de kilometerprijs berekend, zonder rekening te houden met de afschrijving, zoals reeds is opgemerkt.

Omdat voor diverse merken de afschrijving nogal verschilt is dat moeilijk in dit programma in te bouwen. Er zijn nu eenmaal bepaalde merken die sterker in prijs dalen dan andere merken. Voor de werkelijke afschrijving kunt U het best uitgaan van de tabellen, die door de ANWB worden samengesteld.

In het voorbeeld op de volgende pagina ziet U een fictieve opgave voor een auto van merk X.

M A A N D O U E R Z I C H T

MAAND Februari			t/m Februari
BELASTING : f	245.00	f	245.00
BRANDSTOF : f	175.00	f	350.00
ONDERHOUD : f	175.00	f	175.00
VERZEKERING: f	0.00	f	0.00
REPERATIES : f	0.00	f	96.75

MAANDTOTAAL: f 595.00 f 866.75

M A A N D O U E R Z I C H T

MAAND Juli			t/m Juli
BELASTING : f	0.00	f	245.00
BRANDSTOF : f	175.00	f	1400.00
ONDERHOUD : f	0.00	f	425.00
VERZEKERING: f	875.00	f	1085.00
REPERATIES : f	125.00	f	331.75

MAANDTOTAAL: f 1175.00 f 3486.75

J A A R O U E R Z I C H T

BELASTING : f	245.00
BRANDSTOF : f	1400.00
ONDERHOUD : f	425.00
VERZEKERING: f	1085.00
REPERATIES : f	331.75

JAARTOTAAL : f 3486.75
GEREDEN Kms.: 8000

KILOMETERPRIJS: f 0.44

```

10 REM *****
   **
   **  A U T O K O S T E N  **
   ** ----- **
   *****
20 REM ** (C) 1985 R-J.DONKERS **
   *****
30 CLS:WIDTH 40:KEYOFF:COLOR 15,1,1
40 CLEAR 4000
50 DIM M$(12),K$(12,5),K(12,5),S(5),T(12
),TT(12)
60 REM *****
   **  M$=MAANDEN ,K=KOSTEN  **
   **  K$=KOSTENSOORTEN    **
   **  T=MAANDTOTAAL       **
   *****
70 GOSUB 2000
100 REM *****
   **  H O O F D M E N U  **
   *****
110 CLS:PRINT SPC(10);"AUTOKOSTEN"
120 PRINT SPC(9);"===== "
130 LOCATE 5,4:PRINT"1. GEGEVENS INVOERE
N"
140 LOCATE 5,6:PRINT"2. GEGEVENS TOEVOEG
EN"
150 LOCATE 5,8:PRINT"3. GEGEVENS OPSLAAN
"
160 LOCATE 5,10:PRINT"4. GEGEVENS INLEZE
N"
170 LOCATE 5,12:PRINT"5. GEGEVENS PRINTE
N (S/P)"
180 LOCATE 5,14:PRINT"6. JAAROVERZICHT (
S/P)"
190 LOCATE 5,16:PRINT"7. STOPPEN MET PRO
GRAMMA"
200 LOCATE 8,22:PRINT"TOETS UW KEUZE IN"
210 Q$=INKEY$:IF Q$="" THEN 210
220 IF Q$<"1" OR Q$>"7" THEN 210
230 Q=VAL(Q$)

```



```

240 ON Q GOTO 500,800,1600,1800,1000,120
0,2400
300 GOTO 100
500 REM *****
      ** GEGEVENS INVOEREN **
      *****

510 COLOR 15,4,4:CLS
520 IF A THEN 570
530 LOCATE 0,5:PRINT" GEEF DE KM-STAND A
AN HET BEGIN "
540 LOCATE 0,7:INPUT"KM-STAND :";A
550 LOCATE 0,21:PRINT" TOETS >> RETURN <
< VOOR VERVOLG"
560 CLS
570 PRINT SPC(8);"GEGEVENS INVOEREN"
580 PRINT SPC(7);"-----"
590 LOCATE 0,4:INPUT"WELKE MAAND (1/12):
";Q
600 FOR N=1 TO 5
610 LOCATE 0,6+N:PRINT N;". ";K$(Q,N)
620 NEXT N
630 FOR N=1 TO 5
640 LOCATE 17,6+N:INPUT B
650 K(Q,N)=B
660 LOCATE 17,6+N:PRINT USING"F #####.##
";K(Q,N)
670 I(Q)=I(Q)+K(Q,N)
680 NEXT N
700 LOCATE 0,16:PRINT"MAANDTOTAAL ";M$(Q
);": ";USING"F #####.##";I(Q)
740 LOCATE 0,21:PRINT"TOETS >C<=CORRECTI
E >M<=MENU "
750 Q$=INKEY$:IF Q$="" THEN 750
760 IF Q$="C" OR Q$="c" THEN I(Q)=0:FOR
N=1 TO 5:K(Q,N)=0:NEXT N:GOTO 500
770 IF Q$="M" OR Q$="m" THEN 100
780 GOTO 750
800 REM *****
      ** GEGEVENS TOEVOEGEN **
      *****

```

```

810 COLOR 15,4,4:CLS
820 PRINT SPC(8);"GEGEVENS TOEVOEGEN"
830 PRINT SPC(7);"=====
840 LOCATE 0,4:INPUT"VOOR WELKE MAAND (1
/12):";Q
845 FOR N=1 TO 5
850 LOCATE 0,6+N:PRINT N;". ";K$(Q,N)
855 LOCATE 17,6+N:PRINT USING"F #####.##
";K(Q,N)
865 LOCATE 29,6+N:INPUT"+/- ";B
870 K(Q,N)=K(Q,N)+B
875 LOCATE 17,6+N:PRINT USING"F #####.##
";K(Q,N):LOCATE 29,6+N:PRINT SPC(10)
880 T(Q)=T(Q)+B
885 NEXT N
890 LOCATE 0,16:PRINT"MAANDTOTAAL ";M$(Q
);":";USING"F #####.##";T(Q)
895 LOCATE 0,21:PRINT"TOETS >C<=CORRECTI
E >M<=MENU "
900 Q$=INKEY$:IF Q$="" THEN 900
910 IF Q$="C" OR Q$="c" THEN T(Q)=0:FOR
N=1 TO 5:K(Q,N)=0:NEXT N:GOTO 800
920 IF Q$="M" OR Q$="m" THEN 100
930 GOTO 900
1000 REM *****
** GEGEVENS PRINTEN(S/P)**
*****
1010 COLOR 1,11,11:CLS:FOR N=1 TO 5:S(N)
=0:NEXT N:SI=0
1020 PRINT SPC(8);"GEGEVENS PRINTEN"
1030 PRINT SPC(7);"=====
1040 LOCATE 0,4:INPUT"VAN WELKE MAAND WI
LT U DE GEGEVENS (1/12) :";Q
1050 FOR N=1 TO Q
1055 FOR M=1 TO 5
1060 S(M)=S(M)+K(N,M)
1065 NEXT M
1070 NEXT N
1075 LOCATE 0,7:PRINT"WILT U DE MAANDGEG
EVENS : "

```

```

1080 PRINT"          >>> OP HET SCHERM (S
) OF,"
1085 PRINT"          >>> NAAR DE PRINTER (P
)"
1090 LOCATE 0,21:PRINT" TOETS UW KEUZE
(S of P) IN.."
1095 Q$=INKEY$:IF Q$="" THEN 1095
1100 IF Q$="S" OR Q$="s" THEN O$="CRT:":
GOTO 1120
1105 IF Q$="P" OR Q$="p" THEN O$="LPT:":
GOTO 1120
1110 GOTO 1095
1120 CLS
1130 OPEN O$ FOR OUTPUT AS #1
1135 PRINT#1,SPC(8);"M A A N D O U E R Z
I C H T"
1140 PRINT#1,SPC(7);"=====
=====
"
1145 PRINT#1,"MAAND ";M$(Q);"
t/m ";M$(Q)
1150 PRINT#1,
1160 PRINT#1,K$(1,1);USING" F #####.##"
;K(Q,1);:PRINT#1,USING" F #####.##";S
(1)
1165 PRINT#1,K$(1,2);USING" F #####.##"
;K(Q,2);:PRINT#1,USING" F #####.##";S
(2)
1170 PRINT#1,K$(1,3);USING" F #####.##"
;K(Q,3);:PRINT#1,USING" F #####.##";S
(3)
1175 PRINT#1,K$(1,4);USING" F #####.##"
;K(Q,4);:PRINT#1,USING" F #####.##";S
(4)
1180 PRINT#1,K$(1,5);USING" F #####.##"
;K(Q,5);:PRINT#1,USING" F #####.##";S
(5)
1181 PRINT#1,
1182 PRINT#1,"=====
=====
"

```

```

1183 PRINT#1,"MAANDTOTAAL:";USING" F ###
###.##";I(Q);
1184 FOR N=1 TO 5
1185 ST=ST+S(N)
1186 NEXT N
1187 PRINT#1, USING" F #####.##";ST
1188 CLOSE#1
1189 LOCATE 0,21:PRINT" TOETS >> M << U
OOR MENU"
1190 QS=INKEY$:IF QS="" THEN 1190
1195 IF QS="M" OR QS="m" THEN 100
1196 GOTO 1190
1200 REM *****
** JAAROVERZICHT (S/P) **
*****
1210 COLOR 1,11,11:CLS
1215 INPUT"WAT IS KM.-EINDSTAND NU:";B
1216 CLS:FOR Q=1 TO 5:S(Q)=0:NEXT Q
1217 FOR Q=1 TO 12:IT(Q)=0:NEXT Q:JT=0
1220 PRINT SPC(8);"J A A R O V E R Z I C
H T"
1230 PRINT SPC(7);"-----
-----"
1240 LOCATE 0,4:PRINT"WILT U HET JAAROVE
RZICHT : "
1250 LOCATE 0,5:PRINT" >>> OP HET
SCHERM (S) OF,"
1260 LOCATE 0,6:PRINT" >>> NAAR DE
PRINTER (P)"
1270 LOCATE 0,21:PRINT" TOETS UW KEUZE
(S of P) IN.."
1280 QS=INKEY$:IF QS="" THEN 1280
1290 IF QS="S" OR QS="s" THEN OS="CRT.":
GOTO 1320
1300 IF QS="P" OR QS="p" THEN OS="LPT.":
GOTO 1320
1310 GOTO 1280
1320 CLS
1330 FOR N=1 TO 12
1340 FOR M=1 TO 5

```



```

1350 S(M)=S(M)+K(N,M)
1360 NEXT M
1370 IT(N)=IT(N)+I(N)
1380 NEXT N
1390 FOR N=1 TO 12:JT=JT+IT(N):NEXT N
1400 OPEN Q$ FOR OUTPUT AS #1
1410 PRINT#1,SPC(8);"J A A R O U E R Z I
  C H T"
1420 PRINT#1,SPC(7);"-----
-----"
1430 PRINT#1,
1440 PRINT#1,K$(1,1);USING" F #####.##"
;S(1)
1450 PRINT#1,K$(1,2);USING" F #####.##"
;S(2)
1460 PRINT#1,K$(1,3);USING" F #####.##"
;S(3)
1470 PRINT#1,K$(1,4);USING" F #####.##"
;S(4)
1480 PRINT#1,K$(1,5);USING" F #####.##"
;S(5)
1490 PRINT#1,
1500 PRINT#1,"-----
-----"
1510 PRINT#1,"JAARTOTAAL :";USING" F ###
###.##";JT
1520 PRINT#1,"GEREDEN Kms.:";B-A
1530 G=JT/(B-A)
1540 PRINT#1,
1550 PRINT#1,"KILOMETERPRIJS:";USING" F
###.##";G
1560 PRINT#1,"-----
-----"
1570 CLOSE#1
1580 LOCATE 0,21:PRINT" TOETS >> M << U
OOR MENU"
1590 Q$=INKEY$:IF Q$="" THEN 1590
1595 IF Q$="M" OR Q$="m" THEN 100
1596 GOTO 1590

```



```

1600 REM *****
      **  GEGEVENS OPSLAAN  **
      *****

1610 COLOR 4,7,7:CLS
1620 PRINT SPC(4);"GEGEVENS OPSLAAN"
1630 PRINT SPC(3);"-----"
1640 LOCATE 0,4:PRINT"GEGEVENS OPSLAG OP
      DATACASSETTE "
1650 PRINT"PLAATS EEN LEGE CASSETTE IN R
      ECORDER "
1660 PRINT:PRINT"-----
      -----"
1670 PRINT:PRINT"ONDER WELKE NAAM WILT U
      DE DATA OP TAPE ZETTEN ? (MAX.6 Letters
      )."
1680 PRINT:INPUT"NAAM: ";P$
1690 PRINT:PRINT"ZET RECORDER OP OPNEMEN
      EN TOETS DAARNA >> RETURN <<"
1700 Q$=INKEY$:IF Q$="" THEN 1700
1710 IF ASC(Q$)<>13 THEN 1700
1720 OPEN "CAS:P$" FOR OUTPUT AS#1
1725 PRINT#1,A
1730 FOR N=1 TO 12
1740 FOR M=1 TO 5
1750 PRINT#1,K(N,M)
1760 NEXT M
1765 PRINT#1,I(N)
1770 NEXT N
1775 CLOSE#1
1780 LOCATE 0,21:PRINT" TOETS >> M << U
      OOR MENU"
1785 Q$=INKEY$:IF Q$="" THEN 1785
1790 IF Q$="M" OR Q$="m" THEN 100
1795 GOTO 1785
1800 REM *****
      **  GEGEVENS INLEZEN  **
      *****

1810 COLOR 4,7,7:CLS
1820 PRINT SPC(4);"GEGEVENS INLEZEN"
1830 PRINT SPC(3);"-----"

```

```

1840 LOCATE 0,4:PRINT"GEGEVENS INLEZEN U
AN DATACASSETTE "
1850 PRINT"PLAATS DE JUISTE CASSETTE IN
RECORDER "
1860 PRINT:PRINT"-----
-----"
1870 PRINT:PRINT"ONDER WELKE NAAM WILT U
DE DATA VAN TAPE INLEZEN? (MAX.6 Letter
s). "
1880 PRINT:INPUT"NAAM: ";P$
1890 PRINT:PRINT"ZET RECORDER OP PLAY EN
TOETS DAARNA >> RETURN <<"
1900 Q$=INKEY$:IF Q$="" THEN 1900
1910 IF ASC(Q$)<>13 THEN 1900
1920 OPEN "CAS:P$" FOR INPUT AS#1
1925 INPUT#1,A
1930 FOR N=1 TO 12
1940 FOR M=1 TO 5
1950 INPUT#1,K(N,M)
1960 NEXT M
1965 INPUT#1,T(N)
1970 NEXT N
1975 CLOSE#1
1980 LOCATE 0,21:PRINT" TOETS >> M << U
OOR MENU"
1985 Q$=INKEY$:IF Q$="" THEN 1985
1990 IF Q$="M" OR Q$="m" THEN 100
1995 GOTO 1985
2000 REM *****
** INITIALISEREN **
*****
2010 M$(1)="Januari ":M$(2)="Februari "
:M$(3)="Maart ":M$(4)="April ":M$(
5)="Mei ":M$(6)="Juni ":M$(7)="
Juli "
2020 M$(8)="Augustus ":M$(9)="September"
:M$(10)="Oktober ":M$(11)="November ":M
$(12)="December "
2030 FOR N=1 TO 12
2040 RESTORE 2200

```

```

2050 FOR M=1 TO 5
2060 READ A$
2070 K$(N,M)=A$
2080 NEXT M
2090 NEXT N
2100 RETURN
2200 DATA "BELASTING :", "BRANDSTOF :",
"ONDERHOUD :", "VERZEKERING:", "REPERATIE
S : "
2400 REM *****
** PROGRAMMA BEEINDIGEN **
*****

2410 CLS
2420 PRINT " U WILT HET PROGRAMMA STOPPEN
."
2430 LOCATE 0,3:PRINT "VOORDAT WE ECHT ST
OPPEN , WIL IK VAN U WETEN OF U DE GEGE
VENS AL OPGESLAGEN HEEFT ."
2440 PRINT:PRINT "ALS ALLES IN ORDE IS TO
ETST U > E < "
2450 PRINT:PRINT "WILT U NIET STOPPEN TOE
TST U DAN > M < "
2460 LOCATE 0,21:PRINT " TOETS UW KEUZE (
E of M) IN..."
2470 Q$=INKEY$: IF Q$="" THEN 2470
2480 IF Q$="E" OR Q$="e" THEN END
2490 IF Q$="M" OR Q$="m" THEN 100
2500 GOTO 2470

```

MUSIC-O-THEEK

Dit is een bestandsprogramma voor grammofoonplaten, Musicassettes of Compact Discs.

Het programma is ideaal voor het ontsluiten van zgn. verzamel-LP's. Hoe vaak is het U al niet gebeurd, dat U wist dat U een bepaald muziekstuk in Uw discotheek had, maar niet wist op welke plaat dat ook al weer stond.

Voordat U nu Uw hele collectie moet doorzoeken, kunt U met behulp van dit programma snel een bepaald stuk localiseren.

U kunt namelijk elke song of stuk apart invoeren en omdat er dan verwezen wordt naar het plaatnummer heeft U desbetreffende song snel teruggevonden. Nooit meer zoeken dus.

Per File kunt U 300 verschillende songs en muziekstukken opslaan. Als dat voor Uw collectie niet voldoende is, kunt U Uw collectie ook in genres indelen en dan per genre een file aanmaken.

Bijv.: een file Klassiek,
POP,
Gospel,
Country & Western,
Folk,
etc.etc.

Bij het invoeren van gegevens geeft U de Naam van de song, de componist, genre, tijdsduur en plaatnummer.

U kunt op alle ingevoerde gegevens terugzoeken.

Het programma is volledig menugestuurd en werkt met INKEY\$ voor optimaal bedieningsgemak. Verder wordt elke routine van het programma voorafgegaan door een REM-statement zodat U een duidelijk inzicht in het programma krijgt.


```

10 REM *****
    ** M U S I C - O - T H E E K**
    **          (C) 1985          **
    **   ROBERT-JAN DONKERS      **
    *****

15 CLEAR 9000
20 DIM N$(300),C$(300),S$(300),O$(300),I
  (300)
30 P=0
40 CLS:WIDTH 40:COLOR 15,4,4:KEYOFF
50 PRINT"  T E R M I N A L  >> MUZIEKIND
EX <<"
60 PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
  ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^"

70 LOCATE 3,3:PRINT" <1> INVOEREN VAN GE
GEVENS"
80 LOCATE 3,5:PRINT" <2> ZOEKEN VAN GEGE
VENS"
90 LOCATE 3,7:PRINT" <3> WIJZIGEN VAN GE
GEVENS"
100 LOCATE 3,9:PRINT" <4> RAADPLEGEN VAN
  BESTAND"
110 LOCATE 3,11:PRINT" <5> BESTAND NAAR
  PRINTER"
120 LOCATE 3,13:PRINT" <6> BESTAND VAN T
  APE INLEZEN"
130 LOCATE 3,15:PRINT" <7> BESTAND OP TA
  PE VASTLEGGEN"
140 LOCATE 3,17:PRINT" <8> NIEUW BESTAND
  CREEREN"
150 LOCATE 3,19:PRINT" <9> PROGRAMMA EIN
  DIGEN"
160 PRINT:PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
  ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^"

170 PRINT"      T O E T S   UW  KEUZE  IN
  !!"
180 Q$=INKEY$:IF Q$="" THEN 180
190 IF Q$<"1" OR Q$>"9" THEN 180
200 Q=VAL(Q$)

```



```

560 LOCATE 0,10:PRINT" <3> OP MUZIEKSTUK
"
570 LOCATE 0,12:PRINT" <4> OP GENRE"
580 LOCATE 0,14:PRINT" <5> OP TIJD"
590 LOCATE 0,16:PRINT" <6> OP COMPONIST+
MUZIEKSTUK"
600 LOCATE 0,18:PRINT" <7> TERUG NAAR HO
OFDMENU"
610 Q$=INKEY$:IF Q$="" THEN 610
620 IF Q$<"1" OR Q$>"7" THEN 610
630 Q=VAL(Q$)
640 ON Q GOTO 2000,2100,2200,2300,2400,2
500,40
700 REM *****
** WIJZIGEN VAN GEGEVENS **
*****
710 CLS:PRINT"W I J Z I G E N VAN GEGEV
ENS"
720 PRINT"~~~~~
~~~~~":PRINT
730 PRINT"ALLE GEGEVENS MOETEN OPNIEUW W
ORDEN INGEVULD."
740 INPUT"WELK VOLGNUMMER MOET GEWIJZIGD
WORDEN :";Q
750 IF Q>P THEN PRINT"DIT NUMMER ZIT NIE
T IN HET BESTAND !!":FOR N=1 TO 500:NEXT
N:GOTO 710
760 Z$="ER STAAT :":Y$="WIJZIGEN IN "
765 PRINT"P L A A T N U M M E R:"
770 PRINT Z$;N$(Q)
780 PRINT Y$;:INPUT N$(Q)
785 PRINT"C O M P O N I S T I : "
790 PRINT Z$;C$(Q)
800 PRINT Y$;:INPUT C$(Q)
805 PRINT"M U Z I E K S T U K : "
810 PRINT Z$;S$(Q)
820 PRINT Y$;:INPUT S$(Q)
830 PRINT Z$;O$(Q)
835 PRINT"G E N R E : "
840 PRINT Y$;:INPUT O$(Q)

```

```

850 PRINT Z$;T(Q)
855 PRINT" T I J D : "
860 PRINT Y$;:INPUT T(Q)
870 PRINT:PRINT"NOG MEER WIJZIGEN < J/N
>"
880 Q$=INKEY$:IF Q$="" THEN 880
885 IF Q$="J" OR Q$="j" THEN 700
890 IF Q$="N" OR Q$="n" THEN 40
895 GOTO 880
900 REM *****
** RAADPLEGEN VAN BESTAND **
*****
910 CLS:PRINT"R A A D P L E G E N VAN H
ET BESTAND"
920 PRINT"IN HET BESTAND ZIJN ";P;" STUK
KEN ":PRINT"OPGENOMEN.":FOR N=1 TO 500:
NEXT N
930 FOR I=1 TO P
940 GOSUB 1000
950 NEXT I
960 PRINT:PRINT"EINDE VAN BESTAND.":PRIN
T".....
^"
970 PRINT" TOETS >> RETURN << VOOR MENU"
980 Q$=INKEY$:IF Q$="" THEN 980
985 IF ASC(Q$)=13 THEN 40
990 GOTO 980
1000 CLS:PRINT"VOLGno: ";I
1005 PRINT:PRINT" PLAATno: ";N$(I)
1010 PRINT"COMPONIST: ";C$(I)
1020 PRINT" TITEL: ";S$(I)
1030 PRINT" GENRE: ";O$(I)
1040 PRINT" TIJD: ";T(I)
1050 LOCATE 0,17:PRINT"TOETS >>RETURN<<
VOOR VERVOLG":PRINT" >M< IS TERUG NAA
R HOOFDMENU"
1060 Q$=INKEY$:IF Q$="" THEN 1060
1070 IF ASC(Q$)=13 THEN RETURN
1080 IF Q$="M" OR Q$="m" THEN 40
1090 GOTO 1060

```

```

1100 REM*****
      **          PRINTROUTINE          **
      *****
1110 CLS:PRINT" P R I N T E N VAN BESTA
ND"
1120 PRINT:PRINT"ZET UW PRINTER KLAAR EN
TOETS DAARNA >> RETURN << VOOR PRIN
TEN OF >M<= MENU"
1130 Q$=INKEY$:IF Q$="" THEN 1130
1140 IF ASC(Q$)=13 THEN 1200
1150 IF Q$="M" OR Q$="m" THEN 40
1160 GOTO 1130
1200 LPRINT"MUZIEK INDEX"
1205 LPRINT
1210 FOR I=1 TO P
1220 LPRINT"VOLGNUMMER :";I
1230 LPRINT"PLAATNUMMER :";N$(I)
1240 LPRINT"COMPONIST :";C$(I)
1250 LPRINT"MUZIEKSTUK :";S$(I)
1260 LPRINT"GENRE :";O$(I)
1270 LPRINT"TIJD IN MIN.:";T(I)
1280 LPRINT
1290 NEXT I
1295 GOTO 40
1300 REM*****
      ** GEGEVENS VAN TAPE INLEZEN**
      *****
1310 CLS:PRINT"WELKE FILE WILT U INLEZEN
???" :INPUT F$:IF LEN(F$)>6 THEN 1310
1320 PRINT:PRINT"ZET UW RECORDER KLAAR E
N PLAATS JUISTE CASSETTE IN DE RECORDER
!!"
1330 PRINT:PRINT"TOETS DAARNA >> RETURN
<< (>M<= MENU)"
1340 Q$=INKEY$:IF Q$="" THEN 1340
1350 IF ASC(Q$)=13 THEN 1400
1360 IF Q$="M" OR Q$="m" THEN 40
1370 GOTO 1340
1400 OPEN"CAS:F$" FOR INPUT AS#1
1410 INPUT#1,P

```



```

1420 FOR I=1 TO P
1430 INPUT#1,N$(I)
1440 INPUT#1,C$(I)
1450 INPUT#1,S$(I)
1460 INPUT#1,O$(I)
1470 INPUT#1,T(I)
1480 NEXT I
1490 GOTO 40
1500 REM*****
      **      GEGEVENS OP TAPE      **
      **      VASTLEGGEN            **
      *****
1510 CLS:PRINT"ONDER WELKE NAAM WILT U D
E FILE OPSLAAN":INPUT F$:IF LEN(F$)>6 TH
EN 1510
1520 PRINT:PRINT"ZET UW RECORDER KLAAR E
N PLAATS JUISTE CASSETTE IN DE RECORDER
!!"
1530 PRINT:PRINT"TOETS DAARNA >> RETURN
<< (>M<= MENU)"
1540 Q$=INKEY$:IF Q$="" THEN 1540
1550 IF ASC(Q$)=13 THEN 1600
1560 IF Q$="M" OR Q$="m" THEN 40
1570 GOTO 1540
1600 OPEN"CAS:F$" FOR OUTPUT AS#1
1610 PRINT#1,P
1620 FOR I=1 TO P
1630 PRINT#1,N$(I)
1640 PRINT#1,C$(I)
1650 PRINT#1,S$(I)
1660 PRINT#1,O$(I)
1670 PRINT#1,T(I)
1680 NEXT I
1690 GOTO 40
1700 REM*****
      **BEGINNEN MET NIEUW BESTAND**
      *****
1710 CLS:PRINT"BEGINNEN MET N I E U W
BESTAND"

```



```

1720 PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
1730 PRINT:PRINT"ALS U HET HUIDIGE BESTA
ND NIET MEER - 2 NODIG HEEFT, OF U HEEFT
HET REEDS OP CASSETTE VASTGELEGD, DA
N TOETST U >> RETURN << ,ANDERS >
M< VOOR MENU."
1740 Q$=INKEY$:IF Q$="" THEN 1740
1750 IF ASC(Q$)-13 THEN RUN
1760 IF Q$="M" OR Q$="m" THEN 40
1770 GOTO 1740
1900 REM*****
** EINDIGEN VAN PROGRAMMA **
*****
1910 CLS:PRINT"PROGRAMMA WORDT BEEINDIGD
!!"
1920 PRINT"HEEFT U DE GEGEVENS AL OP TAP
E GEZET ?"
1930 PRINT:PRINT"ZO JA, DAN TOETST U RET
URN !"
1940 PRINT:PRINT"ZO NIET, TOETS DAN >M<
VOOR TERUGKEER 72 NAAR HOOFDMENU !"
1950 Q$=INKEY$:IF Q$="" THEN 1950
1960 IF ASC(Q$)=13 THEN END
1970 IF Q$="M" THEN 40
1980 GOTO 1950
2000 REM*****
** ZOEKEN VAN GEGEVENS IN **
** HET TOTALE BESTAND **
*****
2010 REM*****
** ZOEKEN OP PLAATNUMMER **
*****
2020 CLS:PRINT"Z O E K E N OP PLAATNUMM
ER"
2030 PRINT:INPUT"WELK PLAATNUMMER: ";Z$
2040 FOR I=1 TO P
2050 IF N$(I)=Z$ THEN GOSUB 1000
2060 NEXT I
2070 GOTO 3000

```

```

2100 REM*****
      **      ZOEKEN OP COMPONIST      **
      *****
2110 CLS:PRINT"Z O E K E N  OP COMPONIST
"
2120 PRINT:INPUT"GEEF NAAM COMPONIST:";Z
$
2130 FOR I=1 TO P
2140 IF C$(I)=-Z$ THEN GOSUB 1000
2150 NEXT I
2160 GOTO 3000
2200 REM*****
      **      ZOEKEN OP MUZIEKSTUK      **
      *****
2210 CLS:PRINT"Z O E K E N  OP MUZIEKSTU
K"
2220 PRINT:INPUT"GEEF NAAM MUZIEKSTUK:";
Z$
2230 FOR I=1 TO P
2240 IF S$(I)=-Z$ THEN GOSUB 1000
2250 NEXT I
2260 GOTO 3000
2300 REM*****
      **      ZOEKEN OP GENRE      **
      *****
2310 CLS:PRINT"Z O E K E N  OP GENRE"
2320 PRINT:INPUT"WELK GENRE:";Z$
2330 FOR I=1 TO P
2340 IF O$(I)=-Z$ THEN GOSUB 1000
2350 NEXT I
2360 GOTO 3000
2400 REM*****
      **      ZOEKEN OP TIJD      **
      *****
2410 CLS:PRINT"Z O E K E N  OP TIJD"
2420 PRINT:PRINT"DE TE ZOEKEN TIJD WORDT
      BEGRENSD DOOR      EEN MINIMUM EN EEN MAXI
      MUM.                DEZE MIN. EN MAX. MOET
      U OPGEVEN."
2430 PRINT:INPUT"MAX.TIJD IN MINUTEN:";M
A

```

```

2440 INPUT"MIN.TIJD IN MINUTEN:";MI
2450 FOR I=1 TO P
2460 IF T(I)<MA AND T(I)>=MI THEN GOSUB
  1000
2470 NEXT I
2480 GOTO 3000
2500 REM*****
      **          ZOEKEN OP          **
      **  COMPONIST + MUZIEKSTUK  **
      *****
2510 CLS:PRINT"Z O E K E N  OP COMPONIST
  + MUZIEKSTUK"
2520 PRINT:INPUT"GEEF NAAM COMPONIST:";Z
  $
2525 INPUT"GEEF NAAM MUZIEKSTUK:";Y$
2530 FOR I=1 TO P
2540 IF C$(I)=Z$ AND S$(I)=Y$ THEN GOSUB
  1000
2550 NEXT I
3000 PRINT:PRINT"EINDE BESTAND !"
3010 PRINT"NOG MEER ZOEKEN ? < J/N > "
3020 Q$=INKEY$:IF Q$="" THEN 3020
3030 IF Q$="J" OR Q$="j" THEN 500
3040 IF Q$="N" OR Q$="n" THEN 40
3050 GOTO 3020

```

LEDENADMINISTRATIE

Dit programma is ideaal voor allerlei verenigingen die de registratie van hun leden willen vereenvoudigen.

Het programma wordt volledig menu-gestuurd. Het hoofdmenu staat boven aan het scherm en geeft U de volgende mogelijkheden:

- CURSOR '>' vooruit 'bladeren' door het bestand.
- CURSOR '<' terug 'bladeren' in het bestand.
- TOETS 'U' Veranderen van gegevens van desbetreffend lid. Alles opnieuw invoeren.
- TOETS 'L' Nieuw lid. U krijgt automatisch het volgende lidnummer en U kunt daarna de gegevens invullen.
- TOETS 'N' Op lidnummer zoeken.
De cursor springt naar de positie direct achter 'LIDNo:'. U geeft hier een lidnummer en direct daarna verschijnen de gegevens van dat lid op het scherm, als dat nummer tenminste bestaat.
- TOETS 'P' Printen. Hier komt U in de print-routine, U heeft 2 mogelijkheden: De gegevens voluit printen of alleen een adresstrook.
Tevens bepaalt U aan de hand van de index welke gegevens geprint moeten worden.
- TOETS 'S' Het ledenbestand op tape of disk zetten.
- TOETS 'T' Het ledenbestand van tape of disk inlezen.

Het gebruik van de INDEX.

U heeft twee indexmogelijkheden tot Uw beschikking, waarmee U een onderscheidt kunt maken tussen de diverse leden.

Voor een sportvereniging zouden de volgende indexen gebruikt kunnen worden:

Index 1 wordt gebruikt voor het soort lid:

dit kan zijn: DJR = Dames Junior lid
DSR = Dames Senior lid
HJR = Heren Junior lid
HSR = Heren Senior lid
VET = Veteraan lid
DON = Donateur
BES = Bestuurslid

Index 2 kan gebruikt worden om aan te geven in welke maand het lidmaatschap moet worden voldaan. (Als dit per jaar geschiedt, bij maandelijkse contributie is dit overbodig).

Bijv. JAN, APR, AUG etc.etc. of om aan te geven in welk jaar het lidmaatschap is ingegaan; bijv. 83,84,85 etc.etc.

Door nu consequent de door U vastgestelde indexen te gebruiken kunt U op eenvoudige wijze adreslabels aanmaken voor bijvoorbeeld het versturen van invitaties voor een clubavond. Eenmaal per jaar een brief verzenden aan de donateurs is nu een koud kunstje. De mogelijkheden zijn legio en ik wens U dan ook veel gemak van dit programma.

```

10 REM *****
**      LEDENADMINISTRATIE      **
** MET OPTIE VOOR PRINTEN **
**      VAN ADRESLABELS      **
*****

20 WIDTH 40:CLS:COLOR 15,4,4:KEYOFF
30 CLEAR 11000:M=100:DIM D$(M)
40 Z$=SPACES(40)
50 Y$=SPACES(25)
60 LOCATE 0,19:PRINT"NIEUWE DATA:"
70 LOCATE 0,21:INPUT"VANAF LID No: "; I
80 IF I=0 THEN 70
90 F=I:N=I:C=0:II=I
1000 REM *****
**                                     **
**      H O O F D M E N U      **
**                                     **
*****

1010 CLS:COLOR 1,10,10
1020 LOCATE 1,5:PRINT"LIDNo:"
1021 LOCATE 1,7:PRINT"Naam : "
1022 LOCATE 1,9:PRINT"Tel. : "
1023 LOCATE 1,11:PRINT"Adres:"
1024 LOCATE 6,12:PRINT": "
1025 LOCATE 6,13:PRINT": "
1026 LOCATE 6,14:PRINT": "
1027 LOCATE 6,15:PRINT": "
1028 LOCATE 0,17:PRINT"Index1:"
1029 LOCATE 0,18:PRINT"Index2:"
2000 LOCATE 0,0:PRINT"MENU: '>'=VOORUIT Z
OEKEN   '<'=IERUG      -----'U'=VERANDERE
N       'L'=NIEUW LID      'N'=OP NUMMER
ZOEKEN 'P'=PRINTEN      'S'=BESTAND S
AVEN   'T'=INLEZEN"
2010 PRINT STRING$(39,"-")
2020 GOSUB 8000
2030 LOCATE 0,0:PRINT Z$;Z$;Z$;Z$;Z$
2040 IF Q$="N" THEN 2500
2050 IF ASC(Q$)=29 THEN 3100
2060 IF ASC(Q$)=28 THEN 3200

```

```

2070 IF QS="U" THEN 4000
2080 IF QS="L" THEN 4100
2090 IF QS="P" THEN 5000
2100 IF QS="S" THEN 6000
2110 IF QS="I" THEN 7000
2120 GOTO 2000
2500 REM *****
      **      ZOEKEN OP LIDNUMMER      **
      *****
2505 IF D$(1)=" " THEN LOCATE 1,20:PRINT"
      BESTAND IS LEEG !":FOR L=1 TO 1000:NEXT
      L:LOCATE 1,20:PRINT Z$:GOTO 2000
2510 LOCATE 8,5:INPUT I
3000 IF I<F OR I>=N THEN FOR L=7 TO 18:LO
      CATE 7,L:PRINT Y$:NEXT L:GOTO 2000
3010 C=I
3020 GOTO 3300
3100 IF D$(1)=" " THEN LOCATE 1,20:PRINT"
      BESTAND IS LEEG !":FOR L=1 TO 1000:NEXT
      L:LOCATE 1,20:PRINT Z$:GOTO 2000
3105 IF C>=N-1 THEN 2000
3110 C=C+1
3120 GOTO 3300
3200 IF D$(1)=" " THEN LOCATE 1,20:PRINT"
      BESTAND IS LEEG !":FOR L=1 TO 1000:NEXT
      L:LOCATE 1,20:PRINT Z$:GOTO 2000
3205 IF C<=F THEN 2000
3210 C=C-1
3300 FOR A= 5 TO 18
3310 LOCATE 7,A:PRINT Y$
3320 NEXT A
3330 LOCATE 8,5:PRINT C
3340 P=7
3350 FOR L=7 TO 15
3360 IF L=8 OR L=10 THEN L=L+1
3370 LOCATE 8,L:
3380 IF P>100 THEN 3500
3390 Q=ASC(MID$(D$(C+1-F),P,1))
3400 P=P+1
3410 IF Q>127 THEN 3440

```

```

3420 PRINT CHR$(Q);
3430 GOTO 3380
3440 PRINT CHR$(Q-128);
3450 NEXT L
3500 LOCATE 8,17:PRINT LEFT$(D$(C+1-F),3
)
3510 LOCATE 8,18:PRINT MID$(D$(C+1-F),4,
3)
3520 GOTO 2000
4000 REM *****
**      GEGEVENS VERANDEREN      **
*****

4010 IF C<F OR C>=N THEN 2000
4020 GOTO 4200
4100 REM *****
**      GEGEVENS NIEUW LID      **
*****

4105 IF N<F+M THEN 4130
4110 LOCATE 0,21:PRINT"      FILE IS VOL
!!!!"
4115 FOR Q=1 TO 1000:NEXT Q:LOCATE 0,21:
PRINT Z$
4120 GOTO 2000
4130 C=N
4140 N=N+1
4150 FOR A=5 TO 18
4160 LOCATE 7,A:PRINT Y$
4170 NEXT A
4200 LOCATE 0,21:PRINT Z$:LOCATE 8,5:PRI
NT C
4210 P=7:D$(C+1-F)="      "
4220 FOR L=7 TO 15
4230 IF L=8 OR L=10 THEN L=L+1
4240 LOCATE 7,L:
4250 J$=Y$:I$=""
4260 IF P>100 THEN 4370
4270 INPUT I$
4280 IF LEN(I$)>24 THEN I$=LEFT$(I$,24)
4290 IF I$="" THEN I$=""
4300 K=LEN(I$)

```



```

4310 IF P+K>100 THEN I$=LEFT$(I$,101-P)
4320 K=LEN(I$)
4330 MID$(J$,2,K+1)=I$
4335 C$=MID$(I$,K,1)
4340 MID$(I$,K,1)=CHR$(ASC(C$)+128)
4350 D$(C+1-F)=D$(C+1-F)+I$
4360 P=P+K
4370 LOCATE 7,L:PRINT J$
4380 NEXT L
4400 LOCATE 7,17:INPUT I$
4420 MID$(D$(C+1-F),1,3)=I$
4430 LOCATE 7,18:INPUT I$
4440 MID$(D$(C+1-F),4,3)=I$
4450 GOTO 2000
5000 REM *****
      ** PRINTEN MET DE OPTIE OM **
      ** ADRESLABELS AF TE DRUKKEN**
      *****
5005 IF D$(1)="" THEN LOCATE 1,20:PRINT"
  BESTAND IS LEEG !":FOR L=1 TO 1000:NEXT
  L:LOCATE 1,20:PRINT 2$:GOTO 2000
5010 CLS:COLOR 15,4,4
5020 PRINT"AAN WELKE INDEX VOORWAARDEN M
  OETEN DE  ADRESSEN VOLDOEN ?"
5030 PRINT STRING$(39,"-")
5040 PRINT"VERSTREK DE VOLGENDE GEGEVENS
  : "
5050 LOCATE 0,5:INPUT"INDEX1: ";I$
5055 X$=I$
5060 IF LEN(X$)>3 THEN 5050
5065 IF I$<>" " THEN I$=X$
5070 LOCATE 8,5:PRINT I$
5080 LOCATE 0,7:INPUT"INDEX2: ";J$
5085 X$=J$
5090 IF LEN(X$)>3 THEN 5080
5095 IF J$<>" " THEN J$=X$
5100 LOCATE 8,7:PRINT J$
5110 LOCATE 0,9:INPUT"VOLUIT (J/N) (J=VO
  LLEDIG N=ADRESLABEL): ";K$
5120 IF K$="" THEN 5110

```

```

5130 IF LEN(K$)>1 THEN K$=LEFT$(K$,1)
5140 IF K$="J" OR K$<>"j" OR K$="N" OR K
$="n" THEN 5200 ELSE 5110
5200 LPRINT
5210 FOR C=F TO N-1
5220 L$=D$(C-F+1)
5230 IF I$<>" " AND I$<>LEFT$(L$,3) THEN
5390
5240 IF J$<>" " AND J$<>MID$(L$,4,3) THEN
5390
5250 IF K$="N" THEN 5280
5260 LPRINT C;"--";LEFT$(L$,6);"--"
5280 P=7
5290 FOR L=1 TO 7
5300 IF K$="N" AND L=2 THEN 5380
5305 IF P>100 THEN 5370
5310 Q=ASC(MID$(L$,P,1))
5315 P=P+1
5320 IF Q>127 THEN 5360
5330 LPRINT CHR$(Q);
5340 GOTO 5310
5360 LPRINT CHR$(Q-128);
5370 LPRINT
5380 NEXT L
5390 NEXT C
5400 LPRINT
5410 GOTO 1000
6000 REM *****
** BESTAND OPSLAAN **
** OP TAPE OF DISKETTE **
*****

6010 CLS:COLOR 15,4,4
6020 PRINT:PRINT"WILT U OPSLAAN OP TAPE
OF DISKETTE ?"
6030 PRINT:INPUT"(T) of (D) ";Q$
6040 PRINT:PRINT
6050 INPUT"ONDER WELKE NAAM MOET DE FILE
WORDEN OPGESLAGEN:";K$
6060 IF LEN(K$)>6 THEN GOTO 6010
6080 IF Q$="T" THEN GOTO 6100
6085 IF Q$="D" THEN GOTO 6200
6090 GOTO 6010

```

```

6100 MOTOR ON
6110 OPEN "CAS:K$" FOR OUTPUT AS#1
6120 GOTO 6300
6200 OPEN K$ FOR OUTPUT AS#1
6300 NN=N-II+1
6305 PRINT#1,II
6306 PRINT#1,NN
6310 FOR Y=1 TO NN
6320 PRINT #1,D$(Y)
6330 NEXT Y
6340 CLOSE #1
6350 IF Q$="I" THEN MOTOR OFF
6360 GOTO 1000
7000 REM *****
      **          BESTAND INLEZEN          **
      **          VAN TAPE OF DISKETTE     **
      *****

7010 CLS:COLOR 15,4,4
7020 PRINT:PRINT"WILT U INLEZEN VAN TAPE
      OF DISKETTE ?"
7030 PRINT:INPUT"(I) of (D) ";Q$
7040 PRINT:PRINT
7050 INPUT"ONDER WELKE NAAM MOET DE FILE
      WORDEN INGELEZEN:";K$
7060 IF LEN(K$)>6 THEN GOTO 7010
7080 IF Q$="I" THEN GOTO 7100
7085 IF Q$="D" THEN GOTO 7200
7090 GOTO 7010
7100 MOTOR ON
7110 OPEN "CAS:K$" FOR INPUT AS#1
7120 GOTO 7300
7200 OPEN K$ FOR INPUT AS#1
7300 INPUT#1,II
7305 INPUT#1,NN
7310 FOR Y=1 TO NN
7320 INPUT #1,D$(Y)
7330 NEXT Y
7340 CLOSE #1
7350 IF Q$="I" THEN MOTOR OFF
7360 N=NN+II-1
7370 GOTO 1000
8000 Q$=INKEY$:IF Q$="" THEN 8000
8010 RETURN

```

OMREKENEN ANGELSAKSISCH NAAR METRIEK STELSEL

Als U voor beroep of hobby ook engelstalige literatuur leest, bent U waarschijnlijk al dikwijls geconfronteerd met het angelsaksisch stelsel van maten en gewichten.

Vaak kunt U door ervaring wel een benadering geven van de maten en gewichten in het ons bekende metrieke stelsel. Maar een nauwkeurige berekening is vaak een hele klus.

Vooraf voor vertaalwerk uit het engels van bijv. technische literatuur is 't noodzakelijk ook een zo exact mogelijke omrekening te maken van de angelsaksische maten.

Voor lezers van dit boek, die met dit soort problemen geconfronteerd worden is het een uitkomst nu dit programma bij de hand te hebben.

Het Programma is volledig menugestuurd en daardoor eenvoudig te bedienen.

De diverse REM-statements verschaffen U snel een inzicht in de opbouw van het programma.

U kunt met dit programma zowel omrekenen van Angelsaksische- naar Metrieke maten als ook van Metrieke- naar Angelsaksische maten.

U kunt Lengtematen,
 Oppervlakten,
 Gewichten,
 Inhoudsmaten,
 Kubieke inhoudsmaten
 en Temperaturen omrekenen.

Zoals U in de listing kunt zien is er veelvuldig gebruik gemaakt van 'FIX'. Dit is gedaan om snel met de restwaarden van een berekening verder te kunnen rekenen.

Voor de Angelsaksische maten is voor de duidelijkheid de Engelse benaming aangehouden.

Ik hoop dat U zich met dit programma veel rekenwerk zult besparen.


```

10 REM *****
    **      OMREKENEN VAN MATEN,      **
    **      GEWICHTEN EN INHOUDSMATEN**
    **      (C)1985 Robert-Jan Donkers**
    **      voor MSX Computers      **
    *****

20 SCREEN 0:WIDTH 40:COLOR 15,4,4:KEYOFF
30 CLEAR 500
40 DIM A$(3),A(3),B(3),C(4),K(3)
100 REM *****
    **      H O O F D M E N U      **
    *****

110 CLS
120 PRINT"OMREKEN-PROGRAMMA";SPC(19);"MS
X":PRINT STRING$(39,"-"):PRINT
130 PRINT:PRINT:PRINT SPC(16);"M E N U":
PRINT SPC(16);"====="
140 LOCATE 2,9:PRINT"[1]....METRIEK NAA
R ANGELSAKSISCH"
150 LOCATE 2,11:PRINT"[2]....ANGELSAKSI
SCH NAAR METRIEK"
160 LOCATE 6,21:PRINT"TOETS UW KEUZE IN
[1] of [2]"
170 Q$=INKEY$:IF Q$="" THEN 170
180 IF Q$<"1" OR Q$>"2" THEN 170
190 IF Q$="1" THEN LOCATE 0,2:PRINT"METR
IEK NAAR ANGELSAKSISCH":N=1
200 IF Q$="2" THEN LOCATE 0,2:PRINT"ANGE
LSAKSISCH NAAR METRIEK":N=2
210 FOR P=7 TO 21:LOCATE 0,P:PRINT SPACE
$(39):NEXT P
220 LOCATE 2,8:PRINT"[1].....
....LENGTEMATEN"
230 LOCATE 2,10:PRINT"[2].....
OPPERVLAKTEMATEN"
240 LOCATE 2,12:PRINT"[3].....
.....GEWICHTEN"
250 LOCATE 2,14:PRINT"[4].....
....INHOUDSMATEN"

```

```

260 LOCATE 2,16:PRINT"[5].....KUBI
EKE INHOUDSMATEN"
270 LOCATE 2,18:PRINT"[6].....
...TEMPERATUREN"
280 LOCATE 2,20:PRINT"[7].....
.....STOPPEN"
290 LOCATE 5,22:PRINT"TOETS UW KEUZE IN
[1] t/m [7]"
300 Q$=INKEY$:IF Q$="" THEN 300
310 IF Q$<"1" OR Q$>"7" THEN 300
320 Q=VAL(Q$)
330 IF Q=1 THEN P=3:IF N=1 THEN RESTORE
2000 ELSE RESTORE 2100
340 IF Q=2 THEN P=2:IF N=1 THEN RESTORE
2001 ELSE RESTORE 2101
350 IF Q=3 THEN P=2:IF N=1 THEN RESTORE
2002 ELSE RESTORE 2102
360 IF Q=4 THEN P=1:IF N=1 THEN RESTORE
2003 ELSE RESTORE 2103
370 IF Q=5 THEN P=2:IF N=1 THEN RESTORE
2004 ELSE RESTORE 2104
380 IF Q=6 THEN P=1:IF N=1 THEN RESTORE
2005 ELSE RESTORE 2105
390 IF Q=7 THEN CLS:END
400 FOR X=5 TO 22:LOCATE 0,X:PRINT SPACE
$(39):NEXT X
420 FOR R=1 TO P
430 READ A$(R),A(R)
440 NEXT R
450 FOR R=1 TO P
460 LOCATE 5,8+R:PRINT A$(R)
470 NEXT R
480 IF N=1 THEN ON Q GOTO 500,600,700,80
0,900,1000
490 IF N=2 THEN ON Q GOTO 1100,1200,1300
,1400,1500,1600
500 REM *****
* LENGTEMATEN METRIEK>ENGELS *
*****
510 FOR R=1 TO P

```

```

520 LOCATE 25,8+R:INPUT B(R)
525 C(R)=B(R)*A(R)
530 NEXT R
540 K(1)=FIX(C(1))
550 C(2)=(C(1)-K(1))*1760+C(2)
555 K(2)=FIX(C(2))
560 C(3)=(C(2)-K(2))*36+C(3)
565 K(3)=FIX(C(3))
575 LOCATE 5,13:PRINT"is equivalent aan:
"
580 LOCATE 5,15:PRINT USING"MILES #####
";K(1)
585 LOCATE 5,16:PRINT USING"YARDS #####
";K(2)
590 LOCATE 5,17:PRINT USING"INCH #####
";K(3)
595 LOCATE 5,22:PRINT"TOETS [RETURN]"
596 Q$=INKEY$:IF Q$="" THEN 596
597 IF ASC(Q$)=13 THEN 100 ELSE 596
600 REM *****
      * OPPEVLAKTEN METRIEK>ENGELS *
      *****
610 FOR R=1 TO P
620 LOCATE 25,8+R:INPUT B(R)
625 C(R)=B(R)*A(R)
630 NEXT R
640 K(1)=FIX(C(1))
650 C(2)=(C(1)-K(1))*640+C(2)
655 K(2)=FIX(C(2))
675 LOCATE 5,13:PRINT"is equivalent aan:
"
680 LOCATE 5,15:PRINT USING"SQ.MILES ###
###";K(1)
685 LOCATE 5,16:PRINT USING"ACRES     ###
###";K(2)
695 LOCATE 5,22:PRINT"TOETS [RETURN]"
696 Q$=INKEY$:IF Q$="" THEN 696
697 IF ASC(Q$)=13 THEN 100 ELSE 696
700 REM *****
      * GEWICHTEN METRIEK>ENGELS *
      *****

```

```

710 FOR R=1 TO P
720 LOCATE 25,8+R:INPUT B(R)
725 C(R)=B(R)*A(R)
730 NEXT R
740 K(1)=FIX(C(1))
750 C(2)=(C(1)-K(1))*16+C(2)
755 K(2)=FIX(C(2))
775 LOCATE 5,13:PRINT"is equivalent aan:
"
780 LOCATE 5,15:PRINT USING"Lbs      ###
###";K(1)
785 LOCATE 5,16:PRINT USING"Oz      ###
###";K(2)
795 LOCATE 5,22:PRINT"TOETS [RETURN]"
796 Q$=INKEY$:IF Q$="" THEN 796
797 IF ASC(Q$)=13 THEN 100 ELSE 796
800 REM *****
      *   INHOUD      METRIEK>ENGELS   *
      *****

810 FOR R=1 TO P
820 LOCATE 25,8+R:INPUT B(R)
825 C(R)=B(R)*A(R)
830 NEXT R
840 K(1)=FIX(C(1))
850 C(2)=(C(1)-K(1))*32
855 K(2)=FIX(C(2))
875 LOCATE 5,13:PRINT"is equivalent aan:
"
880 LOCATE 5,15:PRINT USING"GALLONS  ###
###";K(1)
885 LOCATE 5,16:PRINT USING"F1 Oz    ###
###";K(2)
895 LOCATE 5,22:PRINT"TOETS [RETURN]"
896 Q$=INKEY$:IF Q$="" THEN 896
897 IF ASC(Q$)=13 THEN 100 ELSE 896
900 REM *****
      *KUBIEKE INHOUD METRIEK>ENGELS *
      *****

910 FOR R=1 TO P
920 LOCATE 25,8+R:INPUT B(R)

```



```

925 C(R)=B(R)*A(R)
930 NEXT R
940 K(1)=FIX(C(1))
950 C(2)=(C(1)-K(1))*46656!+C(2)
955 K(2)=FIX(C(2))
960 K(3)=0
966 IF K(2)>=1728 THEN K(3)=K(3)+1:K(2)=
K(2)-1728:GOTO 966
975 LOCATE 5,13:PRINT"is equivalent aan:
"
980 LOCATE 5,15:PRINT USING"Cub Yard ###
###";K(1)
983 LOCATE 5,16:PRINT USING"Cub Feet ###
###";K(3)
985 LOCATE 5,17:PRINT USING"Cub Inch ###
###";K(2)
995 LOCATE 5,22:PRINT"TOETS [RETURN]"
996 Q$=INKEY$:IF Q$="" THEN 996
997 IF ASC(Q$)=13 THEN 100 ELSE 996
1000 REM*****
      * TEMPERATUREN  METRIEK>ENGELS *
      *****
1010 FOR R=1 TO P
1020 LOCATE 25,8+R:INPUT B(R)
1025 C(R)=B(R)*A(R)
1030 NEXT R
1040 K(1)=C(1)+32
1075 LOCATE 5,13:PRINT"is equivalent aan
:"
1080 LOCATE 5,15:PRINT USING"Fahrenheit#
#####.##";K(1)
1095 LOCATE 5,22:PRINT"TOETS [RETURN]"
1096 Q$=INKEY$:IF Q$="" THEN 1096
1097 IF ASC(Q$)=13 THEN 100 ELSE 1096
1100 REM*****
      * LENGTEMATEN  ENGELS>METRIEK *
      *****
1110 FOR R=1 TO P
1120 LOCATE 25,8+R:INPUT B(R)
1125 C(R)=B(R)*A(R)

```

```

1130 NEXT R
1140 K(1)=FIX(C(1))
1150 C(2)=(C(1)-K(1))*1000+C(2)
1155 K(2)=FIX(C(2))
1160 C(3)=(C(2)-K(2))*100+C(3)
1165 K(3)=FIX(C(3))
1166 IF K(3)>100 THEN K(2)=K(2)+1:K(3)=K
(3)-100:GOTO 1166
1175 LOCATE 5,13:PRINT"is equivalent aan
:"
1180 LOCATE 5,15:PRINT USING"Km      #####
#";K(1)
1185 LOCATE 5,16:PRINT USING"Meter #####
#";K(2)
1190 LOCATE 5,17:PRINT USING"Cm      #####
#";K(3)
1195 LOCATE 5,22:PRINT"TOETS [RETURN]"
1196 Q$=INKEY$:IF Q$="" THEN 1196
1197 IF ASC(Q$)=13 THEN 100 ELSE 1196
1200 REM*****
      * OPPEVLAKTEN  ENGELS>METRIEK *
      *****
1210 FOR R=1 TO P
1220 LOCATE 25,8+R:INPUT B(R)
1225 C(R)=B(R)*A(R)
1230 NEXT R
1240 K(1)=FIX(C(1))
1250 C(2)=(C(1)-K(1))*100+C(2)
1255 K(2)=FIX(C(2))
1266 IF K(2)>=100 THEN K(1)=K(1)+1:K(2)=
K(2)-100:GOTO 1266
1275 LOCATE 5,13:PRINT"is equivalent aan
:"
1280 LOCATE 5,15:PRINT USING"KM 2      ##
####";K(1)
1285 LOCATE 5,16:PRINT USING"Ha      ##
####";K(2)
1295 LOCATE 5,22:PRINT"TOETS [RETURN]"
1296 Q$=INKEY$:IF Q$="" THEN 1296
1297 IF ASC(Q$)=13 THEN 100 ELSE 1296

```

```

1300 REM*****
      *   GEWICHTEN   ENGELS>METRIEK *
      *****
1310 FOR R=1 TO P
1320 LOCATE 25,8+R:INPUT B(R)
1325 C(R)=B(R)*A(R)
1330 NEXT R
1340 K(1)=FIX(C(1))
1350 C(2)=(C(1)-K(1))*1000+C(2)
1355 K(2)=FIX(C(2))
1366 IF K(2)>-1000 THEN K(1)-K(1)+1:K(2)
      =K(2)-1000:GOTO 1366
1375 LOCATE 5,13:PRINT"is equivalent aan
      : "
1380 LOCATE 5,15:PRINT USING"Kg           ##
      ####";K(1)
1385 LOCATE 5,16:PRINT USING"Gr           ##
      ####";K(2)
1395 LOCATE 5,22:PRINT"TOETS [RETURN]"
1396 Q$=INKEY$:IF Q$="" THEN 1396
1397 IF ASC(Q$)=13 THEN 100 ELSE 1396
1400 REM*****
      * INHOUDSMATEN   ENGELS>METRIEK *
      *****
1410 FOR R=1 TO P
1420 LOCATE 25,8+R:INPUT B(R)
1425 C(R)=B(R)*A(R)
1430 NEXT R
1440 K(1)=FIX(C(1))
1450 C(2)=(C(1)-K(1))*1000
1455 K(2)=FIX(C(2))
1475 LOCATE 5,13:PRINT"is equivalent aan
      : "
1480 LOCATE 5,15:PRINT USING"Liters       ##
      ####";K(1)
1485 LOCATE 5,16:PRINT USING"cl/cm3      ##
      ####";K(2)
1495 LOCATE 5,22:PRINT"TOETS [RETURN]"
1496 Q$=INKEY$:IF Q$="" THEN 1496
1497 IF ASC(Q$)=13 THEN 100 ELSE 1496

```

```

1500 REM*****
      *KUBIEKE INHOUD ENGELS>METRIEK *
      *****

1510 FOR R=1 TO P
1520 LOCATE 25,8+R:INPUT B(R)
1525 C(R)=B(R)*A(R)
1530 NEXT R
1540 K(1)=FIX(C(1))
1550 C(2)=(C(1)-K(1))*1000+C(2)
1555 K(2)=FIX(C(2))
1556 C(3)=(C(2)-K(2))*1000
1560 K(3)=FIX(C(3))
1566 IF K(3)>=1000 THEN K(2)=K(2)+1:K(3)
      =K(3)-1000:GOTO 1566
1575 LOCATE 5,13:PRINT"is equivalent aan
:"
1580 LOCATE 5,15:PRINT USING"m3      ##
####";K(1)
1583 LOCATE 5,16:PRINT USING"dm3     ##
####";K(2)
1585 LOCATE 5,17:PRINT USING"cm3     ##
####";K(3)
1595 LOCATE 5,22:PRINT"TOETS [RETURN]"
1596 Q$=INKEY$:IF Q$="" THEN 1596
1597 IF ASC(Q$)=13 THEN 100 ELSE 1596
1600 REM*****
      * TEMPERATUREN ENGELS>METRIEK *
      *****

1610 FOR R=1 TO P
1620 LOCATE 25,8+R:INPUT B(R)
1625 C(R)=(B(R)-32)*A(R)
1630 NEXT R
1640 K(1)=C(1)
1675 LOCATE 5,13:PRINT"is equivalent aan
:"
1680 LOCATE 5,15:PRINT USING"Celsius  #
#####.##";K(1)
1695 LOCATE 5,22:PRINT"TOETS [RETURN]"
1696 Q$=INKEY$:IF Q$="" THEN 1696
1697 IF ASC(Q$)=13 THEN 100 ELSE 1696

```


2000 DATA Km,0.621377,M,1.0936,Cm,0.3937
2001 DATA Km2,0.3861,Ha,2.47096
2002 DATA Kg,2.2046,Gr,0.03572
2003 DATA Lt,0.26418
2004 DATA m3,1.3072,cm3,0.61025
2005 DATA Celsius,1.8
2100 DATA Mile,1.6093,Yards,0.91441,Inch
,2.54
2101 DATA SqrMile,2.590,Acre,0.4047
2102 DATA Lbs,0.4536,Oz,28.3527
2103 DATA Gal,3.7852
2104 DATA Cub.Yard,0.765,Cub.Inch,16.386
2105 DATA Fahrenheit,0.5555555555555556

WEERSTANDKLEURCODES

Met dit programma kunnen knutselaars onder ons op snelle en eenvoudige wijze hun weerstanden sorteren.

Uitgaande van de drie bovenste kleurringen rekent de MSX de waarde van de weerstand voor U uit.

De bediening gebeurt met de cursortoetsen of met joystick.

U krijgt een scherm te zien met onderin een kleurenbalk met daarin alle kleuren die op weerstanden kunnen voor komen.

U kiest eerst de kleur van de eerste ring door het pijltje boven die kleur te zetten en daarna de spatiebalk resp. de vuurknop in te drukken.

Dit doet U ook voor de tweede en derde ring en direct daarna verschijnt de waarde van de weerstand in beeld.

U ziet het eenvoudig en sneller dan met een tabelletje knoeien.

```

10 REM *****
** WEERSTAND-KLEURCODES **
** BEREKENEN **
** VERSIE 1.0 (C)1985 **
** ROBERT-JAN DONKERS **
*****

20 SCREEN0:CLS:KEYOFF
30 PRINT"GEbruikt u JOYSTICK (J/N)"
40 LOCATE 0,21:PRINT" TOETS UW KEUZE IN
(J/N)"
50 QS=INKEY$:IF QS="" THEN 50
60 IF QS="J" OR QS="j" THEN N=1:GOTO 90
70 IF QS="N" OR QS="n" THEN N=0:GOTO 90
80 GOTO 50
90 SCREEN 2:COLOR 15,5,5:CLS
100 X=8:Y=140:R=0
110 SPRITE$(1)=CHR$(16)+CHR$(16)+CHR$(16
)+CHR$(16)+CHR$(254)+CHR$(124)+CHR$(56)+
CHR$(16)
120 PUT SPRITE 1,(X,Y),1,1
130 C=5
140 PRESET(7,7),C
150 OPEN "GRP:"FOR OUTPUT AS#1
160 PRINT#1,"TERMINAL WEERSTANDKLEURCODE
"

170 LINE (0,151)-(23,191),1,BF
180 LINE (24,151)-(47,191),13,BF
190 LINE (48,151)-(71,191),6,BF
200 LINE (72,151)-(95,191),9,BF
210 LINE (96,151)-(118,191),10,BF
220 LINE (119,151)-(143,191),2,BF
230 LINE (144,151)-(167,191),4,BF
240 LINE (168,151)-(191,191),4,BF
250 LINE (168,151)-(191,191),6,B
260 LINE (170,151)-(170,191),6
270 LINE (172,151)-(172,191),6
280 LINE (174,151)-(174,191),6
290 LINE (176,151)-(176,191),6
300 LINE (178,151)-(178,191),6
310 LINE (180,151)-(180,191),6

```

```

320 LINE (182,151)-(182,191),6
330 LINE (184,151)-(184,191),6
340 LINE (186,151)-(186,191),6
350 LINE (188,151)-(188,191),6
360 LINE (192,151)-(215,191),14,BF
370 LINE (216,151)-(239,191),15,BF
380 PRESET(7,55),C:PRINT#1,"1e RING:"
390 PRESET(7,71),C:PRINT#1,"2e RING:"
400 PRESET(7,87),C:PRINT#1,"3e RING:"
410 PRESET(15,111),C:PRINT#1,"WAARDE:"
420 REM *****
      **      HOOFDPROGRAMMA      **
      *****
430 IF R=3 THEN 770
440 STRIG(N) ON
450 IF R<>3 THEN P=STICK(N) ELSE 770
460 IF P=0 THEN X=X:Y=140
470 IF P=3 THEN X=X+24:IF X>224 THEN X=2
480 IF P=7 THEN X=X-24:IF X<8 THEN X=8
490 PUT SPRITE 1,(X,Y),1,1
500 ON STRIG GOSUB 530
510 FOR P=1 TO 100:NEXT P:GOTO 450
520 GOTO 520
530 REM *****
      ** DE KLEURCODES VERTALEN **
      *****
540 R=R+1:X1=87:X2=127:Y1=39:Y2=47
550 IF X=8 THEN C=1:W=0
560 IF X=32 THEN C=13:W=1
570 IF X=56 THEN C=6:W=2
580 IF X=80 THEN C=9:W=3
590 IF X=104 THEN C=10:W=4
600 IF X=128 THEN C=2:W=5
610 IF X=152 THEN C=4:W=6
620 IF X=176 THEN C=4:C1=6:W=7
630 IF X=200 THEN C=14:W=8
640 IF X=224 THEN C=15:W=9
650 IF C1=6 THEN 680
660 LINE (X1,Y1+R*16)-(X2,Y2+R*16),C,BF

```



```

670 GOTO 730
680 LINE (X1,Y1+R*16)-(X2,Y2+R*16),C,BF
690 LINE (X1,Y1+R*16+2)-(X2,Y1+R*16+2),C
1
700 LINE (X1,Y1+R*16+4)-(X2,Y1+R*16+4),C
1
710 LINE (X1,Y1+R*16+6)-(X2,Y1+R*16+6),C
1
720 C1=0
730 IF R=1 THEN T=W
740 IF R=2 THEN T=10*T+W
750 IF R=3 THEN T=T*(10^W)
760 RETURN
770 REM *****
      ** WAARDE VAN DE WEERSTAND**
      **          INVULLEN          **
      *****

780 R=0
790 PRESET (87,111),5
800 PRINT#1,T;CHR$(234)
810 PRESET (0,135),5
820 PRINT#1,"TOETS 'V'=VERVOLG,'S'=STOP"
830 Q$=INKEY$:IF Q$="" THEN 830
840 IF Q$="V" OR Q$="v" THEN 870
850 IF Q$="S" OR Q$="s" THEN END
860 GOTO 830
870 LINE (87,55)-(127,63),5,BF
880 LINE (87,71)-(127,79),5,BF
890 LINE (87,87)-(127,95),5,BF
900 LINE (87,111)-(255,119),5,BF
910 LINE (0,135)-(255,143),5,BF
920 GOTO 420

```

ELEMENTENLIJST

Elementenlijst is een programma dat eenmaal ingetoetst en op tape gezet een ideaal hulpmiddel kan zijn voor scholieren, die scheikunde in hun pakket hebben.

Telkens als U gegevens nodig heeft van een bepaald element hoeft U niet meer te bladeren in een of ander tabellenboek maar een eenvoudige 'druk op de knop' is dan voldoende om de gegevens voor U op het scherm te zetten.

Uiteraard moet U dan wel eerst Uw MSX klaar zetten voor gebruik, voor dat U met het huiswerk begint.

U kunt zoeken op Naam van het element, op symbool of op atoomnummer.

Het is even werken om het programma in te toetsen, maar daarna kunt U er toch veel gemak van hebben.

```

10 REM *****
    **          ELEMENTENLIJST          **
    ** (c) 1985 ROBERT-JAN DONKERS **
    *****
15 REM *****
    **          INITIALISEREN          **
    *****

20 DIM$(106):DIM S$(106):DIM A(106)
25 RESTORE
30 FOR N=1 TO 106
40 READ E$(N):READ S$(N):READ A(N)
50 NEXT N
60 COLOR 15,1,1:KEYOFF:WIDTH 40
70 CLS
80 PRINT "ELEMENTENLIJST (c) 1985 TERMIN
AL "
90 LOCATE 5,5:PRINT " M E N U "
100 LOCATE 5,8:PRINT "1. ZOEKEN OP NAAM
    ELEMENT"
110 LOCATE 5,10:PRINT"2. ZOEKEN OP SYMBO
    OL VAN ELEMENT"
130 LOCATE 5,12:PRINT "3. ZOEKEN OP ATOO
    MNUMBER"
150 LOCATE 5,14:PRINT "4. COMPLETE ELEME
    NTENLIJST"
170 LOCATE 5,16:PRINT "5. STOPPEN"
180 LOCATE 0,21:PRINT " > TOETS UW KEU
    ZE IN < "
190 Q$=INKEY$:IF Q$="" THEN 190
200 IF Q$<"1" OR Q$>"5" THEN 190
210 A=VAL(Q$)
220 ON A GOTO 240,420,620,740,1040
230 GOTO 70
240 REM*****
    ** ZOEKEN OP NAAM VAN ELEMENT **
    *****

245 CLS:POKE 64683!,255
250 PRINT" E L E M E N T E N L I J S T "

```

```

260 PRINT"~~~~~"
~~~~~"
270 PRINT
280 PRINT"  OPZOEKEN OP NAAM."
290 LOCATE 3,6:INPUT"GEEF NAAM ELEMENT :
";A$
320 FOR N=1 TO 106
330 IF A$=E$(N) THEN 360
340 NEXT N
350 LOCATE 0,21:PRINT " ELEMENT BESTAAT
NIET !!!!!!!":FOR P=1 TO 500:NEXT P:GOTO
 245
360 PRINT:PRINT
365 PRINT"NAAM          :";E$(N)
370 PRINT"ATOOMNO.     :";N
375 PRINT"SYMBOOL      :";S$(N)
380 PRINT"MASSA        :";A(N)
400 C=2
410 GOTO 950
420 REM*****
      **      ZOEKEN OP SYMBOOL      **
      *****
425 CLS:POKE 64683!,0
430 PRINT" E L E M E N T E N L I J S T "
440 PRINT"~~~~~"
~~~~~"
450 PRINT:PRINT" OPZOEKEN OP SYMBOOL."
460 PRINT
470 INPUT "GEEF SYMBOOL :";A$
510 FOR N=1 TO 106
520 IF A$=S$(N) THEN 560
530 NEXT N
540 LOCATE 0,21:PRINT " ELEMENT BESTAAT
NIET !!!!!!!"
550 FOR Q=1 TO 500:NEXT Q:GOTO 420
560 PRINT:PRINT
565 PRINT"NAAM          :";E$(N)
570 PRINT"ATOOMNO.     :";N
575 PRINT"SYMBOOL      :";S$(N)
580 PRINT"MASSA        :";A(N)

```



```

600 C=4
610 GOTO 950
620 REM*****
      **   ZOEKEN OP ATOOMNUMMER   **
      *****

625 CLS
630 PRINT" E L E M E N T E N L I J S T "
640 PRINT:PRINT
650 PRINT"OPZOEKEN OP ATOOMNUMMER."
660 LOCATE 3,5:INPUT"GEEF ATOOMNUMMER:";
N
670 IF N<1 OR N>106 THEN LOCATE 0,21:PRI
NT" ELEMENT BESTAAT NIET !!!!!!!!!!!":FO
R P=1 TO 500:NEXT P:GOTO 620
680 PRINT:PRINT
685 PRINT"NAAM           :";E$(N)
690 PRINT"ATOOMNO.      :";N
695 PRINT"SYMBOL        :";S$(N)
700 PRINT"MASSA         :";A(N)
710 C=6
720 GOTO950
740 REM*****
      ** COMPLETE ELEMENTENLIJST **
      *****

741 CLS:PRINT"COMPLETE ELEMENTENLIJST":L
OCATE 0,3:PRINT" NAAR (S)CHERM OF (P)RIN
TER "
742 LOCATE 0,21:PRINT"TOETS UW KEUZE IN
(S) OF (P) "
744 QS=INKEY$:IF QS="" THEN 744
745 IF QS="S" OR QS="s" THEN 748
746 IF QS="P" OR QS="p" THEN 870
747 GOTO 744
748 N=1
750 CLS:PRINT" E L E M E N T E N L I J S
T "
760 LOCATE 0,3:PRINT"COMPLETE LIJST VAN
DE ELEMENTEN ."
780 LOCATE 0,5:PRINT"NAAM           :";E$(N
)

```

```

790 LOCATE 0,7:PRINT"ATOOM-NR.      :";N
800 LOCATE 0,9:PRINT"SYMBOL      :";S$(N
)
810 LOCATE 0,11:PRINT"MASSAGETAL  :";A(N
);"      "
820 LOCATE 0,21:PRINT" TOETS      >RETURN<=
VERVOLG, >M<=MENU      "
830 Q$=INKEY$:IF Q$="-" THEN 830
840 IF ASC(Q$)-13 THEN N=N+1:GOTO850
845 IF Q$="M" OR Q$="m" THEN 70
846 GOTO 830
850 IF N=107 THEN 70
860 GOTO 750
870 REM*****
      ** COMPLETE LIJST NAAR PRINTER **
      *****
880 LPRINT" COMPLETE ELEMENTENLIJST ":LP
RINT
890 FOR N=1 TO 106
900 LPRINT"NAAM              :";E$(N)
910 LPRINT"ATOOM-NR        :";N
920 LPRINT"SYMBOL          :";S$(N)
930 LPRINT"MASSAGETAL     :";A(N)
940 LPRINT:LPRINT:NEXT N
945 GOTO 70
950 LOCATE 0,21:PRINT"          TOETS  >M<=M
ENU    >V<=VERVOLG      "
955 POKE 64683!,255
960 Q$=INKEY$:IF Q$="-" THEN 960
970 IF Q$="M" THEN 70
980 IF Q$="V" THEN 1030
990 GOTO 960
1030 IF C=2 THEN 240
1032 IF C=4 THEN 420
1034 IF C=6 THEN 620
1040 END
2000 DATA "WATERSTOF", "H", 1.00797
2001 DATA "HELIUM", "He", 4.0026
2002 DATA "LITHIUM", "Li", 6.941
2003 DATA "BERYLLIUM", "Be", 9.012

```

2004 DATA "BOOR", "B", 10.811
2005 DATA "KOOLSTOF", "C", 12.0111
2006 DATA "STIKSTOF", "N", 14.0076
2007 DATA "ZUURSTOF", "O", 15.9994
2008 DATA "FLUOR", "F", 18.9984
2009 DATA "NEON", "Ne", 20.179
2010 DATA "NATRIUM", "Na", 22.9898
2011 DATA "MAGNESIUM", "Mg", 24.305
2012 DATA "ALUMINIUM", "Al", 26.9815
2013 DATA "SILICIUM", "Si", 28.086
2014 DATA "FOSFOR", "P", 30.9738
2015 DATA "ZWAVEL", "S", 32.064
2016 DATA "CHLOOR", "Cl", 35.453
2017 DATA "ARGON", "Ar", 39.0948
2018 DATA "KALIUM", "K", 39.102
2019 DATA "CALCIUM", "Ca", 40.08
2020 DATA "SCANDIUM", "Sc", 44.956
2021 DATA "TITAAN", "Ti", 47.90
2022 DATA "VANADIUM", "V", 50.942
2023 DATA "CHROOM", "Cr", 51.996
2024 DATA "MANGAAN", "Mn", 54.938
2025 DATA "IJZER", "Fe", 55.847
2026 DATA "KOBALT", "Co", 58.9332
2027 DATA "NIKKEL", "Ni", 58.71
2028 DATA "KOPER", "Cu", 63.546
2029 DATA "ZINK", "Zn", 64.38
2030 DATA "GALLIUM", "Ga", 69.72
2031 DATA "GERMANIUM", "Ge", 72.59
2032 DATA "ARSEEN", "As", 74.9216
2033 DATA "SELENIUM", "Se", 78.96
2034 DATA "BROOM", "Br", 79.904
2035 DATA "KRYPTON", "Kr", 83.80
2036 DATA "RUBIDIUM", "Rb", 85.47
2037 DATA "STRONTIUM", "Sr", 87.62
2038 DATA "YTTRIUM", "Y", 88.905
2039 DATA "ZIRKONIUM", "Zr", 91.22
2040 DATA "NIOBIUM", "Nb", 92.906
2041 DATA "MOLYBDEEN", "Mo", 95.94
2042 DATA "TECHNETIUM", "Tc", 97
2043 DATA "RUTHENIUM", "Ru", 101.07

2044 DATA "RHODIUM", "Rh", 102.905
 2045 DATA "PALLADIUM", "Pd", 106.4
 2046 DATA "ZILVER", "Ag", 107.870
 2047 DATA "CADMIUM", "Cd", 112.40
 2048 DATA "INDIUM", "In", 114.82
 2049 DATA "TIN", "Sn", 118.69
 2050 DATA "ANTIMOON", "Sb", 121.75
 2051 DATA "TELLUUR", "Te", 127.60
 2052 DATA "JOOD", "I", 126.9045
 2053 DATA "XENON", "Xe", 131.3
 2054 DATA "CESIUM", "Cs", 132.905
 2055 DATA "BARIUM", "Ba", 137.34
 2056 DATA "LANTHAAN", "La", 138.91
 2057 DATA "CERIUM", "Ce", 140.12
 2058 DATA "PRASEODYMIUM", "Pr", 140.907
 2059 DATA "NEODYMIUM", "Nd", 144.24
 2060 DATA "PROMETHIUM", "Pm", 145
 2061 DATA "SAMARIUM", "Sm", 150.35
 2062 DATA "EUROPIUM", "Eu", 151.96
 2063 DATA "GADOLINIUM", "Gd", 157.25
 2064 DATA "TERBIUM", "Tb", 158.924
 2065 DATA "DYSPROSIUM", "Dy", 162.50
 2066 DATA "HOLMIUM", "Ho", 164.930
 2067 DATA "ERBIUM", "Er", 167.26
 2068 DATA "THULIUM", "Tm", 168.934
 2069 DATA "YTTERBIUM", "Yb", 173.04
 2070 DATA "LUTETIUM", "Lu", 174.97
 2071 DATA "HAFNIUM", "Hf", 178.49
 2072 DATA "TANTIAAL", "Ta", 180.948
 2073 DATA "WOLFRAM", "W", 183.85
 2074 DATA "RENIUM", "Re", 186.2
 2075 DATA "OSMIUM", "Os", 190.2
 2076 DATA "IRIDIUM", "Ir", 192.2
 2077 DATA "PLATINA", "Pt", 195.09
 2078 DATA "GOUD", "Au", 196.967
 2079 DATA "KWIK", "Hg", 200.59
 2080 DATA "THALLIUM", "Th", 204.37
 2081 DATA "LOOD", "Pb", 207
 2082 DATA "BISMUTH", "Bi", 208.981
 2083 DATA "POLONIUM", "Po", 209

2084 DATA "ASTAAT", "At", 210
2085 DATA "RADON", "Rn", 222
2086 DATA "FRANCIUM", "Fr", 223
2087 DATA "RADIUM", "Ra", 226
2088 DATA "ACTINIUM", "Ac", 227
2089 DATA "THORIUM", "Th", 232.038
2090 DATA "PROTACTINIUM", "Pr", 140.907
2091 DATA "URANIUM", "U", 238.03
2092 DATA "NEPTUNIUM", "Np", 237
2093 DATA "PLUTONIUM", "Pu", 244
2094 DATA "AMERICIUM", "Am", 243
2095 DATA "CURIUM", "Cm", 247
2096 DATA "BERKELIUM", "Bk", 247
2097 DATA "CALIFORNIUM", "Cf", 251
2098 DATA "EINSTEINIUM", "Es", 252
2099 DATA "FERMIUM", "Fm", 257
2100 DATA "MENDELEVIUM", "Md", 257
2101 DATA "NOBELIUM", "No", 255
2102 DATA "LAWRENCIUM", "Lr", 257
2103 DATA "KURCHATOVIIUM", "Ku", 259
2104 DATA "HANNIUM", "Ha", 262
2105 DATA "UNNILHEXIUM", "Unh", 263

HI-RES TEKENEN

Met dit tekenprogramma kunt U allerlei prachtige schermen ontwerpen.

Er zijn een aantal standaardmogelijkheden ingebouwd o.a.: Cirkels, Rechthoeken, Lijnen, invullen met een kleur, kleur veranderen.

Het programma werkt zowel met Joystick als Cursortoetsen.

Als U het programma opstart komt U in screen 2 en ziet U in het midden een pijlcursor.

Deze kunt U bewegen met cursortoets of joystick. Door tijdens het bewegen gelijktijdig spatiebalk resp. vuurknop in te drukken tekent U een lijn in de kleur die U gekozen heeft.

Verdere bediening:

Als de cursor in beeld is heeft U naast het hierboven beschreven bewegen en tekenen de volgende mogelijkheden:

- TOETS 'F'= Fill routine; een vlak inkleuren
- TOETS 'P'= Oorspronkelijke kleur(=wit=15) instellen.
- TOETS 'K'= Kleur kiezen (15 is de standaardkleur bij het opstarten).
- TOETS 'W'= Wissen.(Hier wordt de kleur van de te wissen lijn hetzelfde als de achtergrond:kleur 4).
- TOETS 'C'= Cirkel tekenen met een door U te bepalen straal r en als middelpunt de huidige cursorpositie.
- TOETS 'L'= Een lijn tussen twee punten.
- TOETS 'B'= Een rechthoek(=Box) tekenen.
- TOETS 'S'= Een screendump op papier.

De diverse mogelijkheden nader toegelicht.

FILL ROUTINE:

U zet de cursor in het te kleuren vlak en druk daarna de 'F'-toets in.

Zorg er voor dat wanneer U een bepaald vlak wilt inkleuren met Kleur C, de lijnen van de figuur ononderbroken zijn en dat deze lijnen van dezelfde kleur C moeten zijn, anders overschrijft de fill-routine Uw gehele tekening.

KLEUR INSTELLEN:

U kunt hiermee de kleuren 1 t/m 15 instellen om daarmee dan een figuur te tekenen. Kies NOOIT kleur 4 want dat is de kleur van de achtergrond en dus wordt Uw tekening onzichtbaar. Met P stelt U de oorspronkelijke kleur weer in. (=wit=15)

Geef de waarde van de kleur als 2 cijfers : 4 wordt dus 04 en 10 blijft 10.

WISSEN:

Hierbij wordt de kleur van het tekenen gelijk aan de achtergrond, zodat wanneer U over een figuur tekent deze gewist wordt. Heeft U alles gewist wat er gewist moest worden, dan stelt U met 'P' de oorspronkelijke kleur weer in.

CIRKELS TEKENEN:

U zet eerst de cursor op het punt waar het middelpunt van de cirkel moet liggen.

Daarna toetst U 'C' in en kunt U de straal van de cirkel invoeren. De waarde van de straal moet U met drie cijfers invoeren.

Moet de straal 40 zijn, dan toetst U 040 in, bij een straal van 100 is dat 100 etc.etc.

LIJNEN TEKENEN:

U zet eerst de cursor op het punt waar de lijn moet beginnen. Zet daar een punt door op spatiebalk of vuurknop te drukken.

Toets dan 'L' om aan te geven dat U een lijn wilt trekken. Verplaats dan de cursor naar het eindpunt van de lijn. Het eindpunt van de lijn wordt door U aangegeven door op de spatiebalk of de vuurknop te drukken. Dan wordt de lijn tussen deze twee punten getrokken in een kleur die U met de optie 'K' heeft aangegeven.

RECHTHOEK TEKENEN:

Deze routine maakt gebruik van de mogelijk van MSX-BASIC om een zgn 'BOX' te tekenen.

U geeft eerst aan waar de linkerbovenhoek van de rechthoek komt te liggen. Eerst de cursor daar naar toe verplaatsen, dan de vuurknop of spatiebalk indrukken. Dan toetst U 'B' om aan te geven dat U een rechthoek wilt tekenen.

Verplaats nu de cursor naar het punt waar de rechterbenedenhoek van de rechthoek moet komen te liggen. Druk dan spatiebalk of vuurknop in en U ziet dat de rechthoek wordt getekend.

SCREENDUMP:

Toets 'S' brengt U in een screendump programma, waarmee U de tekening naar de printer stuurt.

Er is in dit programma geen screendump-routine opgenomen, omdat er al diverse gepubliceerd zijn en het onzin is om er nog een te ontwikkelen als er al enkele goede beschikbaar zijn voor MSX.

MSX Computer Magazine publiceerde o.a. CRT-dump, die U zo achter dit programma kunt plakken.


```

10 REM *****
    **      HIGH RESOLUTION      **
    ** TEKENPROGRAMMA SCREEN 2 **
    ** (C)1985 ROBERT-J.DONKERS **
    *****

20 REM *****
    **      INITIALISEREN      **
    *****

30 C=15:W=0
40 COLOR C,4,4:KEYOFF:SCREEN 0:WIDTH 40
50 PRINT:PRINT"GEBRUIKT U JOYSTICK (J/N)
"
60 QS=INKEY$:IF QS="" THEN 60
70 IF QS="J" OR QS="j" THEN N=1:GOTO 100
80 IF QS="N" OR QS="n" THEN N=0:GOTO100
90 GOTO 60
100 SCREEN 2
110 OPEN"GRP:" FOR OUTPUT AS#1
120 SPRITE$(0)=CHR$(252)+CHR$(192)+CHR$(
160)+CHR$(144)+CHR$(136)+CHR$(132)+CHR$(
2)+CHR$(1)
130 X=127:Y=95
140 PUT SPRITE 0,(X,Y),1,0
150 R=STICK(N)
160 IF R=0 THEN X=X:Y=Y
170 IF R=1 THEN Y=Y-1
180 IF R=2 THEN X=X+1:Y=Y-1
190 IF R=3 THEN X=X+1
200 IF R=4 THEN X=X+1:Y=Y+1
210 IF R=5 THEN Y=Y+1
220 IF R=6 THEN X=X-1:Y=Y+1
230 IF R=7 THEN X=X-1
240 IF R=8 THEN X=X-1:Y=Y-1
250 IF X>255 THEN X=255
260 IF X<0 THEN X=0
270 IF Y>182 THEN Y=182
280 IF Y<0 THEN Y=0
290 IF STRIG(N) THEN GOSUB 400
300 AS=INKEY$
310 IF AS="F" OR AS="f" THEN 520

```

```

320 IF A$="P" OR A$="p" THEN C=15
330 IF A$="K" OR A$="k" THEN 430
340 IF A$="W" OR A$="w" THEN C=4
350 IF A$="C" OR A$="c" THEN 550
360 IF A$="B" OR A$="b" THEN 650
370 IF A$="L" OR A$="l" THEN 860
380 IF A$="S" OR A$="s" THEN 1070:REM
    EEN SCREEN-DUMP PROGRAMMA.
390 GOTO 140
400 REM *****
    ** PUNT ZETTEN IN KLEUR C **
    *****

410 PSET(X,Y),C
420 RETURN
430 REM *****
    ** KLEUR C INSTELLEN **
    *****

440 PRESET(0,183),4:PRINT#1,"Welke Kleur
    (1 t/m 15) ":C$=""
450 Q$=INKEY$:IF Q$="" THEN 450
460 IF Q$<"0" OR Q$>"9" THEN 450
470 C$=C$+Q$:IF LEN(C$)<>2 THEN 450
480 C=VAL(C$)
490 FOR Q=1 TO100:NEXT Q
500 LINE (0,183)-(255,191),4,BF
510 GOTO 140
520 REM *****
    ** ULAK INKLEUREN MET C **
    *****

530 PAINT(X,Y),C
540 GOTO 140
550 REM *****
    **CIRKEL TEKENEN MET R EN C**
    *****

560 PRESET(0,183),4:PRINT#1,"Geef straal
    van cirkel ":C$=""
570 Q$=INKEY$:IF Q$="" THEN 570
580 IF Q$<"0" OR Q$>"9" THEN 570
590 C$=C$+Q$:IF LEN(C$)<>3 THEN 570

```

```

600 R=VAL(C$)
610 FOR Q=1 TO100:NEXT Q
620 LINE (0,183)-(255,191),4,BF
630 CIRCLE(X,Y),R,C
640 GOTO 140
650 REM *****
    **   BLOK TEKENEN MET C   **
    *****

660 X1=X:Y1=Y
670 PUT SPRITE 0,(X,Y),1,0
680 R=STICK(N)
690 IF R=0 THEN X=X:Y=Y
700 IF R=1 THEN Y=Y-1
710 IF R=2 THEN X=X+1:Y=Y-1
720 IF R=3 THEN X=X+1
730 IF R=4 THEN X=X+1:Y=Y+1
740 IF R=5 THEN Y=Y+1
750 IF R=6 THEN X=X-1:Y=Y+1
760 IF R=7 THEN X=X-1
770 IF R=8 THEN X=X-1:Y=Y-1
780 IF X>255 THEN X=255
790 IF X<0 THEN X=0
800 IF Y>182 THEN Y=182
810 IF Y<0 THEN Y=0
820 IF NOT STRIG(N) THEN 670
830 IF STRIG(N) THEN GOSUB 400
840 LINE (X1,Y1)-(X,Y),C,B
850 GOTO 140
860 REM *****
    **   LINE TEKENEN MET C   **
    *****

870 X1=X:Y1=Y
880 PUT SPRITE 0,(X,Y),1,0
890 R=STICK(N)
900 IF R=0 THEN X=X:Y=Y
910 IF R=1 THEN Y=Y-1
920 IF R=2 THEN X=X+1:Y=Y-1
930 IF R=3 THEN X=X+1
940 IF R=4 THEN X=X+1:Y=Y+1
950 IF R=5 THEN Y=Y+1
960 IF R=6 THEN X=X-1:Y=Y+1

```

```

970 IF R=7 THEN X=X-1
980 IF R=8 THEN X=X-1:Y=Y-1
990 IF X>255 THEN X=255
1000 IF X<0 THEN X=0
1010 IF Y>182 THEN Y=182
1020 IF Y<0 THEN Y=0
1030 IF NOT STRIG(N) THEN 880
1040 IF STRIG(N) THEN GOSUB 400
1050 LINE (X1,Y1)-(X,Y),C
1060 GOTO 140
1070 REM HIER EEN SCREEN-DUMP PROGRAMMA
1080 REM BIJV. UIT MSX COMPUTER MAGAZINE
1090 REM DAT ER IN DIT BOEK GEEN IS-
      OPGENOMEN KOMT OMDAT ER ENKELE
      GOEDE IN BOVENGENOEMD 'MCM'
      ZIJN GEPUBLICICEERD.
1100 REM DE AUTEUR VINDT HET DAN OOK
      ONZIN OM VOOR DE ZOVEELSTE
      KEER 'HET WIEL WEER UIT TE
      VINDEN' !
1110 REM IN REGEL 380 IS REEDS EEN
      VERWIJZING NAAR DIT SCREENDUMP
      PROGRAMMA OPGENOMEN:TOETS 'S'.
1120 REM EINDIG HET SCREENDUMP-PROG.
      MET EEN REGEL ALS HIERONDER;
      'GOTO 140'.
1130 GOTO 140

```


GRAFIEK PLOTTEN

Met dit programma kunt U bepaalde gegevens Grafisch in beeld brengen.

De gegevens worden per datum geregistreerd. Per datum kunt U max. 3 rubrieken invoeren en in beeld brengen.

Als voorbeeld zou het volgende kunnen:

1. Verkopen
2. Inkopen
3. Kosten

Als U voor deze rubrieken nu de verschillende waarden invoert voor 'n aantal datums ,krijgt U per datum 3 gekleurde balken in beeld, waarvan de lengtes overeenkomen met de door U ingevoerde waarden.

De kop boven de balken waar de diverse waarden ingevuld worden, wordt steeds aangepast aan de hoogste waarde die U heeft ingevoerd. Dit geschiedt in de regel 280, waarbij PP de maximale waarde bijhoudt. De verhoudingen tussen de diverse rubrieken wordt daardoor steeds gegarandeert.

Het programma is volledig menu-gestuurd, waarbij U in het Hoofdmenu de volgende opties heeft:

1. Invoer van gegevens
2. Gegevens opslaan op tape of disk
3. Gegegens inlezen van tape of disk
4. De Grafiek plotten
5. Eindigen met programma

Het programma is ruim voorzien van REM-statements zodat U een duidelijk inzicht krijgt over de diverse routines.

```

10 REM *****
    **      GRAFIEKEN PLOTTEN      **
    **      MET DATUM INDEX        **
    **      VERSIE 1.1 VOOR MSX    **
    ** (C)1985 ROBERT-J.DONKERS   **
    *****

20 SCREEN 0:WIDTH 40:COLOR 15,1,1
30 CLEAR 6000
40 DIM A$(50),A(50,3),P(50,3),B$(3),C(3)
50 C(1)=4:C(2)=6:C(3)=10
60 PP=180:F=1:N=0:X=0:NN=0:KEYOFF
70 Z$=SPACE$(39):Y$=SPACE$(24)
80 GOTO 310
90 REM *****
    **      INVOERROUTINE        **
    *****

100 IF B$(1)="" THEN 110 ELSE 190
110 CLS:PRINT"GRAFIEKEN PLOTTEN MET DATU
MINDEX":PRINT STRING$(39,"*"):PRINT
120 PRINT"U kunt maximaal 3 verschillend
e rubrie- ken invoeren.":PRINT
130 INPUT"hoeveel rubrieken wilt U gebru
iken: ";Q
140 IF Q>3 THEN 90
150 FOR R=1 TO Q
160 PRINT R;"e RUBRIEK:";
170 INPUT B$(R)
180 NEXT R
190 REM *****
    **      INVOEREN VAN DE WAARDEN **
    *****

200 X=NN
210 X=X+1:IF X>50 THEN PRINT"BESTAND IS
VOL !!:FOR J=1 TO 500:NEXT J:GOTO 320
220 CLS:INPUT"DATUM:(ddmmjj)(XXXXXX=EIND
E): ";A$(X)
230 IF A$(X)="XXXXXX" THEN NN=X-1:GOTO 3
10
240 FOR W=1 TO Q
250 PRINT

```

```

260 PRINT:PRINT B$(W);"-";
270 INPUT"Waarde=";A(X,W)
280 IF A(X,W)>PP THEN PP=PP+180:F=F+1:GO
TO 280
290 NEXT W
300 GOTO 210
310 REM *****
      **   H O O F D - M E N U   **
      *****

320 COLOR 15,1,1
330 SCREEN 0:WIDTH 40:CLS:PRINT TAB(16);
" M E N U":PRINT TAB(15);STRING$(9,"-"):P
RINT
340 LOCATE 5,5:PRINT"[1].....INVOEREN
GEGEVENS"
350 LOCATE 5,7:PRINT"[2].....GEGEVENS
OPSLAAN"
360 LOCATE 5,9:PRINT"[3].....GEGEVENS
INLEZEN"
370 LOCATE 5,11:PRINT"[4].....GRAFIEK
EN PLOTTEN"
380 LOCATE 5,13:PRINT"[5].....PROGRAM
MA BEEINDIGEN"
390 LOCATE 5,21:PRINT"Toets Uw keuze in
([1]/[5])"
400 QS=INKEY$:IF QS="" THEN 400
410 IF QS<"1" OR QS>"5" THEN 400
420 QQ=VAL (QS)
430 ON QQ GOTO 90,960,1330,440,1700
440 REM *****
      **   DE GRAFIEKEN PLOTTEN   **
      *****

450 FOR L=1 TO NN
460 FOR M=1 TO Q
470 P(L,M)=55+INT(A(L,M)/F)
480 NEXT M
490 NEXT L
500 SCREEN 2
510 T=0
520 CLS

```

```

530 OPEN"GRP:" FOR OUTPUT AS#1
540 B=55
550 LINE(0,0)-(B,23),15
560 LINE(0,47)-(255,47),15
570 LINE(0,23)-(255,23),15
580 LINE(0,71)-(255,71),15
590 LINE(0,95)-(255,95),15
600 LINE(0,119)-(255,119),15
610 LINE(0,143)-(255,143),15
620 LINE(0,167)-(255,167),15
630 LINE(B,0)-(B,167),15
640 LINE(0,0)-(255,191),15,B
650 PRESET(1,15),4:PRINT#1,"DATA"
660 PRESET(23,2),4:PRINT#1,"TOT"
670 FOR L=1 TO 4
680 K=B+(L*45)
690 LINE(K,12)-(K,23),15 X
700 NEXT L
710 PRESET(87,1),4:PRINT#1,(PP/4)
720 PRESET(128,1),4:PRINT#1,(PP/2)
730 PRESET(171,1),4:PRINT#1,(PP*.75)
740 PRESET(215,1),4:PRINT#1,PP
750 LINE(0,169)-(25,174),C(1),BF
760 LINE(0,176)-(25,182),C(2),BF
770 LINE(0,184)-(25,190),C(3),BF
780 PRESET(32,168),4:PRINT#1,B$(1)
790 PRESET(32,176),4:PRINT#1,B$(2)
800 PRESET(32,184),4:PRINT#1,B$(3)
810 FOR L=1 TO 6
820 D=7+(L*24):T=T+1
830 IF A$(T)="XXXXXX" THEN 910
840 PRESET(2,D),1:PRINT#1,A$(T)
850 G=0+(L*24)
860 FOR M=1 TO Q
870 LINE(B,G)-(P(T,M),G+6),C(M),BF
880 G=G+8
890 NEXT M
900 NEXT L
910 PRESET(136,176),1:PRINT#1,"TOETS [RE
TURN]"

```



```

920 CLOSE#1
930 Q$=INKEY$:IF Q$="" THEN 920
940 IF ASC(Q$)=13 AND T<NN THEN 520
950 IF ASC(Q$)=13 THEN 310 ELSE 930
960 REM *****
    **      GEGEVENS OPSLAAN      **
    *****

970 COLOR15,4,4:CLS
980 PRINT"GEGEVENS OPSLAAN (CASSETTE OF
DISKETTE)":PRINT STRING$(39,"-"):PRINT
990 PRINT"Wilt U de gegevens opslaan op
cassette of diskette?":PRINT
1000 INPUT"(T) of (D) ";Q$
1010 IF Q$<>"T" OR Q$<>"t" THEN IF Q$<>"
D" OR Q$<>"d" THEN 970
1020 PRINT:PRINT
1030 INPUT"Onder welke NAAM moet de FILE
worden opgeslagen :";K$
1040 IF LEN(K$)>6 THEN 1030
1050 IF Q$="T" OR Q$="t" THEN 1070
1060 IF Q$="D" OR Q$="d" THEN 1100
1070 MOTOR ON
1080 OPEN"CAS:K$" FOR OUTPUT AS#1
1090 GOTO 1110
1100 OPEN K$ FOR OUTPUT AS#1
1110 PRINT#1,NN
1120 PRINT#1,Q
1130 PRINT#1,PP
1140 PRINT#1,F
1150 PRINT#1,NN
1160 PRINT#1,Q
1170 FOR M=1 TO NN
1180 PRINT#1,AS(M)
1190 FOR L=1 TO Q
1200 PRINT#1,A(M,Q)
1210 PRINT#1,P(M,Q)
1220 NEXT L
1230 FOR L=1 TO Q
1240 PRINT#1,A(M,Q)
1250 PRINT#1,P(M,Q)

```

```

1260 NEXT L
1270 NEXT M
1280 FOR L=1 TO Q
1290 PRINT#1,B$(L)
1300 NEXT L
1310 IF Q$="T" OR Q$="t" THEN MOTOR OFF
1320 CLOSE#1:GOTO 310
1330 REM *****
      **      GEGEVENS INLEZEN      **
      *****
1340 COLOR15,4,4:CLS
1350 PRINT"GEGEVENS INLEZEN (CASSETTE OF
  DISKETTE)":PRINT STRING$(39,"-"):PRINT
1360 PRINT"Wilt U de gegevens inlezen va
n cassette of diskette ?":PRINT
1370 INPUT"(T) of (D) ";Q$
1380 IF Q$<>"T" OR Q$<>"t" THEN IF Q$<>"
D" OR Q$<>"d" THEN 1340
1390 PRINT:PRINT
1400 INPUT"Onder welke NAAM moet de FILE
  worden ingelezen :";K$
1410 IF LEN(K$)>6 THEN 1400
1420 IF Q$="T" OR Q$="t" THEN 1440
1430 IF Q$="D" OR Q$="d" THEN 1470
1440 MOTOR ON
1450 OPEN"CAS:K$" FOR INPUT AS#1
1460 GOTO 1480
1470 OPEN K$ FOR INPUT AS#1
1480 INPUT#1,NN
1490 INPUT#1,Q
1500 INPUT#1,PP
1510 INPUT#1,F
1520 INPUT#1,NN
1530 INPUT#1,Q
1540 FOR M=1 TO NN
1550 INPUT#1,A$(M)
1560 FOR L=1 TO Q
1570 INPUT#1,A(M,Q)
1580 INPUT#1,P(M,Q)
1590 NEXT L

```

```

1600 FOR L=1 TO Q
1610 INPUT#1,A(M,Q)
1620 INPUT#1,P(M,Q)
1630 NEXT L
1640 NEXT M
1650 FOR L=1 TO Q
1660 INPUT#1,B$(L)
1670 NEXT L
1680 IF Q$="T" OR Q$="t" THEN MOTOR OFF
1690 CLOSE#1:GOTO 310
1700 REM *****
      **      EINDE PROGRAMMA      **
      *****
1710 CLS: PRINT"PROGRAMMA-EINDE":PRINT S
TRING$(39,"-"):PRINT
1720 PRINT"U wilt het programma beeindig
en ?"
1730 PRINT"Heeft U de gegevens al vastge
legd op      Cassette of Diskette ?"
1740 PRINT:PRINT"Als dit NIET nodig is,
of al gebeurd is,toetst U dan [E]=(EINDE
)"
1750 PRINT:PRINT"In alle andere gevallen
toets [RETURN]"
1760 LOCATE 0,21:PRINT"TOETS UW KEUZE IN
: [E] of [RETURN]"
1770 Q$=INKEY$:IF Q$="" THEN 1770
1780 IF Q$="E" OR Q$="e" THEN END
1790 IF ASC(Q$)=13 THEN 310 ELSE 1770

```

L I T E R A T U U R O N D E R Z O E K

Dit programma is , zoals de titel al doet vermoeden, bedoeld om tijdschriftartikelen te ontsluiten.

U kunt aan elk artikel 5 trefwoorden toekennen. Deze trefwoorden zijn het belangrijkste onderdeel van elk literatuuronderzoek.

Gebruik daarom steeds 'standaard' trefwoorden en maak van de gebruikte trefwoorden een lijst; de zgn. THESAURUS. In deze lijst mogen geen twee verschillende trefwoorden met een gelijke betekenis voor komen.

U gebruikt er dan 1 en van het andere wordt dan een verwijzing gemaakt naar het tweede.

Voorbeeld:

Een artikel over programma's voor de MSX computer, kan de volgende trefwoorden krijgen:

1. MSX-computers
2. Programma's
3. Software
4. Microsoft Extended BASIC
5. MSX-Basic

In dit voorbeeld ziet U dat er enkele dubblures in de toekenning van de trefwoorden voor komen.

We spreken voor onszelf af dat we alleen voor ons belangrijke 'standaard' trefwoorden gebruiken.

We kiezen voor: MSX-Computers
 Software
 MSX-Basic

Deze trefwoorden nemen we op in onze lijst van 'standaard'-trefwoorden, de THESAURUS.

De overgebleven trefwoorden die door ons niet als 'standaard' zijn aangemerkt nemen we wel op in onze THESAURUS, maar nu met een verwijzing:

Programma's	ZIE: Software
Microsoft Extended BASIC	ZIE: MSX-Basic

Door op deze wijze consequent alle synoniemen te laten verwijzen naar de door gekozen 'standaard'-trefwoorden, kunnen we voorkomen dat we artikelen niet meer kunnen terugvinden in ons bestand. Tenslotte is het de bedoeling van dit programma literatuuronderzoek dat we alles over een bepaald onderwerp kunnen terugvinden. Bovendien wordt op deze manier een efficiënter gebruik mogelijk van de 5 trefwoorden die ons ter beschikking staan.

In dit programma kunt U bepaalde artikelen alleen opzoeken d.m.v. trefwoorden. Dus opzoeken op auteur of tijdschrift kan alleen door het hele bestand door te 'bladeren'. In het voorbeeld dat hierna is afgedrukt waren drie artikelen ingevoerd. Deze hadden alle drie 1 trefwoord gemeen nl. MSX-Basic. Toen we dus op het trefwoord 'MSX-Basic' gingen zoeken verschenen dan ook al de drie artikelen op papier.

Het programma:

U kunt 100 artikelen invoeren in dit programma zoals dat hier in dit boek is afgedrukt. Hierbij is er rekening mee gehouden dat elk record 180 tekens mag bevatten. De eerste 50 tekens van elk record liggen vast, die zijn gereserveerd voor de 5 trefwoorden. Hieruit volgt dan logischerwijs dat elk trefwoord max. 10 tekens lang mag zijn. In de praktijk blijkt dit voldoende.

Voor de rest van het record blijven dan 130 tekens over, dit blijkt voor de meeste artikelen ruim voldoende te zijn.

In het hierna afgedrukte voorbeeld komen we op:

Record 1: 38 tekens
Record 2: 52 tekens
Record 3: 66 tekens

U ziet dat 130 tekens ruim voldoende is. Door nu afkortingen te gebruiken voor de naam van het tijdschrift (mits U die zelf natuurlijk begrijpt) of de naam inkort, kunt U al veel tekens besparen. Ook jaargangen zoals 1985 zou U kunnen schrijven als 85. Ontdekt U nu dat 130 tekens + 50 voor de trefwoorden voor Uw gebruik echt teveel is, dan kunt U het aantal artikelen uitbreiden.

Als U vindt dat U aan 80 tekens voor het record + 50 voor de trefwoorden voldoende heeft dan zet U de lengte van elk record op 130 tekens. U bespaart daardoor ruimte voor een extra 5000 tekens en kunt U $5000/130 = 38$ artikelen meer opnemen.

U dient dan de volgende aanpassingen te doen:

```
regel 30: M=138  
regel 280: IF N>138 THEN N=138:.....  
regel 340: IF P>130 THEN .....  
regel 980: IF P>130 THEN .....  
regel 1770: IF P>130 THEN .....
```

Verder zijn er geen aanpassingen nodig. Het programma is volledig menu-gestuurd; de keuzes uit het menu werken met INKEY\$ zodat het bedieningsgemak optimaal is. Na een opdracht keert U altijd weer terug naar het hoofdmenu.

Bij het artikelen-overzicht (keuze 3 uit het menu) kunt U door het hele bestand 'bladeren' met behulp van de cursortoetsen '<' & '>'. Verder is het programma voorzien van duidelijke REM-Statements, zodat het programma goed te begrijpen is.

Toepassingen:

Bij studie: voor het verzamelen van literatuur voor studieprojecten, afstudeeropdrachten etc.

Voor hobby: het verzamelen en rangschikken van de relevante literatuur.

Verstandig is het om bij het toepassen van dit programma per onderwerp bijv. MSX Computers een aparte file aan te leggen.

Per ONDERWERP moet U dan wel weer een THESAURUS aanleggen, waar U zich dan strikt aan moet houden om een zo succesvol mogelijk gebruik van dit programma te maken.

LITERATUURONDERZOEK OP TREFWOORD

GEVONDEN LITERATUUR:

Record: 1

TITEL :MSX VEROVERT EUROPA

AUTEUR: BOS.W

TIJDS :RAM

JAAR :1985

No. :10

Blz :18-24

GEVONDEN LITERATUUR:

Record: 2

TITEL :PROGRAMMEER TECHNIEKEN

AUTEUR: DRUIJFF.F

TIJDS :MSX MOZAIEK

JAAR :1985

No. :5

Blz :24-28

GEVONDEN LITERATUUR:

Record: 3

TITEL :WERKEN MET DE VIDEORAM

AUTEUR: IMMERZEEL.M.B.

TIJDS :MSX COMPUTER MAGAZIN

JAAR :1985

No. :2

Blz :40-41


```

10 REM *****
    **      L I T E R A T U U R O N D E R      **
    **      M E T T R E F W O O R D E N - I N D E X      **
    **      V E R S I E 1.0 M S X      **
    ** (C)1985 Robert-J. Donkers      **
    *****

20 WIDTH 40:COLOR 15,1,1:KEYOFF
30 CLEAR 19000:M=100
40 DIM A$(M),T$(3),L1(3)
50 Z$=SPACE$(39):Y$=SPACE$(28)
60 N=1:F=1:C=0:II=1
70 DEFINT L,Q,R,X,Z
80 REM *****
    **      H O O F D M E N U      **
    *****

90 CLS
100 PRINT"L I T E R A T U U R O N D E R
Z O E K":PRINT STRING$(39,"-"):PRINT
110 LOCATE 5,5:PRINT"1. >-----Artikel
invoeren"
120 LOCATE 5,7:PRINT"2. >-----Artikel
afvoeren"
130 LOCATE 5,9:PRINT"3. >-----Artikel
en-overzicht"
140 LOCATE 5,11:PRINT"4. >-----Zoeken
op trefwoorden"
150 LOCATE 5,13:PRINT"5. >-----Litera
tuurlijst Saven"
160 LOCATE 5,15:PRINT"6. >-----Litera
tuurlijst Inlezen"
170 LOCATE 5,17:PRINT"7. >-----Stoppe
n met Programma"
180 LOCATE 5,21:PRINT"Toets Uw Keuze ([1
]/[7]) in !"
190 Q$=INKEY$:IF Q$="" THEN 190
200 IF Q$<"1" OR Q$>"7" THEN 190
210 Q=VAL(Q$)
220 ON Q GOSUB 240,1950,810,1280,2160,24
20,2670
230 GOTO 80

```

```

240 REM *****
      **          INVOER-ROUTINE          **
      *****

250 CLS
260 PRINT"L I T E R A T U R O N D E R
Z O E K":PRINT"INVOERROUTINE":PRINT STRI
NG$(39,"-"):PRINT
270 C=N:N=N+1:A$(C)=STRING$(55," "):P=56
280 IF N>100 THEN N=100:LOCATE 10,10:PRI
NT"BESTAND IS VOL !!!":GOTO 630
290 GOSUB 670
300 FOR L=3 TO 10
310 IF L=4 OR L=6 THEN 470
320 LOCATE 12,L
330 J$=Y$:I$=""
340 IF P>200 THEN 460
350 INPUT I$
360 IF L=3 THEN IF LEN(I$)>50 THEN I$=L
EFT$(I$,50):J$=Y$+Y$
370 IF L=3 THEN 390
380 IF LEN(I$)>20 THEN I$=LEFT$(I$,20)
390 IF I$="" THEN I$=" "
400 K=LEN(I$)
410 MID$(J$,2,K+1)=I$
420 C$=MID$(I$,K,1)
430 MID$(I$,K,1)=CHR$(ASC(C$)+128)
440 A$(C+1-F)=A$(C+1-F)+I$
450 P=P+K
460 LOCATE 12,L:PRINT J$
470 NEXT L
480 FOR L=14 TO 18
490 IF L=14 THEN L1=1:GOTO 510
500 L1=L1+11
510 LOCATE 12,L
520 J$=Y$:I$=""
530 INPUT I$
540 IF LEN(I$)>11 THEN I$=LEFT$(I$,11)
550 IF I$="" THEN I$=" "
560 K=LEN(I$)
570 MID$(J$,2,K+1)=I$

```

```

580 C$=MID$(I$,K,1)
590 MID$(I$,K,1)=CHR$(ASC(C$)+128)
600 MID$(A$(C+1-F),L1,K)=I$
610 LOCATE 12,L:PRINT J$
620 NEXT L
630 LOCATE 0,22:PRINT" [M] = MENU [A]=
NOG EEN ARTIKEL"
640 Q$=INKEY$:IF Q$=""THEN 640
650 IF Q$="M" OR Q$="m" THEN RETURN
660 IF Q$="A" OR Q$="a" THEN 240 ELSE 64
0
670 LOCATE 0,3:PRINT"TITEL      :"
680 LOCATE 0,5:PRINT"AUTEUR(S)  :"
690 LOCATE 0,7:PRINT"TIJDSCHRIFT:"
700 LOCATE 0,8:PRINT"JAAR(GANG)  :"
710 LOCATE 0,9:PRINT"No.         :"
720 LOCATE 0,10:PRINT"Blz.-Blz.   :"
730 LOCATE 0,12:PRINT"TREFWOORDEN:Elk ma
x.10 letters !"
740 LOCATE 0,13:PRINT"-----
-----"
750 LOCATE 0,14:PRINT"1e Trefw.   :"
760 LOCATE 0,15:PRINT"2e Trefw.   :"
770 LOCATE 0,16:PRINT"3e Trefw.   :"
780 LOCATE 0,17:PRINT"4e Trefw.   :"
790 LOCATE 0,18:PRINT"5e Trefw.   :"
800 RETURN
810 REM *****
** ARTIKELEN-OVERZICHT **
*****

820 CLS
830 PRINT"L I T E R A T U U R O N D E R
Z O E K":PRINT"ARTIKELEN-OVERZICHT":PRIN
T STRING$(39,"-"):PRINT
840 IF C=0 THEN LOCATE 10,10:PRINT"BESTA
ND IS LEEG !!!":GOTO 1200
850 R=1
860 IF R<1 THEN R=1
870 IF R>=C+1 THEN 1200
880 GOSUB 670

```

```

890 LOCATE 29,1:PRINT"Record: ";R;" "
900 FOR L=3 TO 18
910 LOCATE 12,L:PRINT Y$
920 NEXT L
930 LOCATE 0,4:PRINT Z$
940 P=56
950 FOR L=3 TO 10
960 IF L=4 OR L=6 THEN 1050
970 LOCATE 12,L
980 IF P>200 THEN 1060
990 Q=ASC(MID$(A$(R),P,1))
1000 P=P+1
1010 IF Q>127 THEN 1040
1020 PRINT CHR$(Q);
1030 GOTO 980
1040 PRINT CHR$(Q-128)
1050 NEXT L
1060 FOR L=14 TO 18
1070 IF L=14 THEN L1=1:GOTO 1120
1080 IF L=15 THEN L1=12
1090 IF L=16 THEN L1=23
1100 IF L=17 THEN L1=34
1110 IF L=18 THEN L1=45
1120 LOCATE 12,L
1130 Q=ASC(MID$(A$(R),L1,1))
1140 L1=L1+1
1150 IF Q>127 THEN 1180
1160 PRINT CHR$(Q);
1170 GOTO 1130
1180 PRINT CHR$(Q-128)
1190 NEXT L
1200 LOCATE 0,21:PRINT"[M]=MENU [>]=DOOR
- [<]=TERUG-BLADEREN"
1210 IF WW=1 THEN LOCATE 10,20:PRINT"[W]
=WISSEN"
1220 Q$=INKEY$:IF Q$=""THEN 1220
1230 IF Q$="M" OR Q$="m" THEN RETURN
1240 IF WW=1 THEN IF Q$="W" OR Q$="w" TH
EN 2070
1250 IF ASC(Q$)=29 THEN R=R-1:GOTO 860

```



```

1260 IF ASC(Q$)=28 THEN R=R+1:GOTO 860
1270 GOTO 1220
1280 REM *****
      ** ZOEKEN OP TREFWOORDEN **
      *****
1290 CLS
1300 PRINT"L I T E R A T U R O N D E R
      Z O E K":PRINT"ZOEKEN OP TREFWOORD":PRI
PRINT STRING$(39,"-"):PRINT
1310 IF C=0 THEN LOCATE 10,10:PRINT"BEST
AND IS LEEG !!!":GOTO 1900
1320 PRINT"U KUNT ZOEKEN OP EEN COMBINAT
IE VAN      DRIE TREFWOORDEN."
1330 PRINT"OP HOEVEEL TREFWOORDEN WILT U
      ZOEKEN ?"
1340 INPUT"AANTAL: ";T:IF T>3 THEN 1280
1350 FOR Y=1 TO T
1360 PRINT Y;:INPUT"e TREFWOORD : ";T$(Y)
1370 LT(Y)=LEN(T$(Y)):IF LT(Y)>11 THEN T
$(Y)=LEFT$(T$(Y),11):LT(Y)=11
1380 H$=CHR$(ASC(RIGHT$(T$(Y),1))+128)
1390 T$(Y)=LEFT$(T$(Y),LT(Y)-1)+H$
1400 NEXT Y
1410 PRINT:PRINT"Wilt U de lijst van gev
onden literatuur op het Scherm of op Pri
nter ?":PRINT
1420 INPUT"[S] of [P] :";Q$
1430 IF Q$="S" OR Q$="s" THEN 1460
1440 IF Q$="P" OR Q$="p" THEN 1470
1450 GOTO 1420
1460 OPEN"CRT:" FOR OUTPUT AS#1:GOTO 148
0
1470 OPEN"LPT:" FOR OUTPUT AS#1
1480 CLS
1490 PRINT#1,"LITERATUURONDERZOEK OP TRE
FWOORD":PRINT#1,STRING$(39,"-")
1500 FOR Z=1 TO C
1510 U=0
1520 FOR X=1 TO T
1530 FOR Q=1 TO S

```

```

1540 B$=""
1550 IF Q=1 THEN L1=1
1560 IF Q=2 THEN L1=12
1570 IF Q=3 THEN L1=23
1580 IF Q=4 THEN L1=34
1590 IF Q=5 THEN L1=45
1600 IF T$(X)=MID$(A$(Z),L1,LT(X)) THEN
U=U+1
1610 NEXT Q
1620 NEXT X
1630 IF U=T THEN GOSUB 1660
1640 NEXT Z
1650 GOTO 1890
1660 CLS:PRINT#1,"GEVONDEN LITERATUUR:"
1670 PRINT#1,"Record:";Z
1680 PRINT#1,"-----"
1690 P=56
1700 FOR L=1 TO 6
1710 IFL=1 THENPRINT#1,"TITEL :";
1720 IFL=2 THENPRINT#1,"AUTEUR:";
1730 IFL=3 THENPRINT#1,"TIJDS :";
1740 IFL=4 THENPRINT#1,"JAAR  :";
1750 IFL=5 THENPRINT#1,"No.   :";
1760 IFL=6 THENPRINT#1,"Blz   :";
1770 IF P>200 THEN 1850
1780 Q=ASC(MID$(A$(Z),P,1))
1790 P=P+1
1800 IF Q>127 THEN 1830
1810 PRINT#1,CHR$(Q);
1820 GOTO 1770
1830 PRINT#1,CHR$(Q-128)
1840 NEXT L
1850 LOCATE 0,21:PRINT"toets [RETURN] vo
or vervolg"
1860 Q$=INKEY$:IF Q$="" THEN 1860
1870 IF ASC(Q$)=13 THEN CLS:LOCATE 0,0:R
ETURN
1880 GOTO 1860
1890 CLOSE#1

```

```

1900 LOCATE 0,21:PRINT"[M] = MENU [Z]=N
OG EEN ONDERZOEK"
1910 Q$=INKEY$:IF Q$=""THEN 1910
1920 IF Q$="M" OR Q$="m" THEN RETURN
1930 IF Q$="Z" OR Q$="z" THEN 1280
1940 GOTO 1910
1950 REM *****
      ** ARTIKEL AFVOEREN/WISSEN**
      *****

1960 CLS
1970 PRINT"L I T E R A T U U R O N D E R
Z O E K":PRINT"ARTIKEL WISSEN/AFVOEREN"
:PRINT STRING$(39,"-"):PRINT
1980 IF C=0 THEN LOCATE 10,10:PRINT"BEST
AND IS LEEG !!!":GOTO 2120
1990 PRINT"Als U een ARTIKEL wilt WISSEN
toets dan [RETURN] , zoniet dan [ESC]"
2000 Q$=INKEY$:IF Q$="" THEN 2000
2010 IF ASC(Q$)=13 THEN WW=1:GOTO 2030
2020 IF ASC(Q$)=27 THEN WW=0:RETURN:GOTO
2000
2030 PRINT:PRINT"Om een artikel te wisse
n gaan we eerst terug naar het literatu
uroverzicht. U zoekt dan hier het ar
tikel dat gewist moet worden en door de
extra optie [W] kunt U dan het gevonden
artikel wissen!"
2040 LOCATE 12,21:PRINT"Toets [RETURN]"
2050 Q$=INKEY$:IF Q$="" THEN 2050
2060 IF ASC(Q$)=13 THEN RETURN ELSE 2050
2070 REM *****
      ** ARTIKEL WISSEN **
      *****

2080 SWAP A$(R),A$(C)
2090 A$(C)=A$(C+1)
2100 C=C-1:N=N-1:WW=0
2110 CLS
2120 LOCATE 0,21:PRINT" [M] = MENU [W]
=Nog 'n ARTIKEL WISSEN"
2130 Q$=INKEY$:IF Q$=""THEN 2130

```

```

2140 IF Q$="M" OR Q$="m" THEN RETURN
2150 IF Q$="W" OR Q$="w" THEN 1950 ELSE
2130
2160 REM *****
      ** LITERATUURLIJST SAVEN **
      *****

2170 CLS
2180 PRINT"L I T E R A T U U R O N D E R
      Z O E K":PRINT"LITERATUURLIJST SAVEN":P
RINT STRING$(39, "-"):PRINT
2190 IF C=0 THEN LOCATE 10,10:PRINT"BEST
AND IS LEEG !!!":GOTO 2390
2200 PRINT"WILT U WEGSCHRIJVEN NAAR TAPE
      OF DISC ?"
2210 PRINT:INPUT"(I) of (D):";Q$
2220 PRINT:PRINT
2230 INPUT"Onder welke NAAM moet de FILE
      worden weggeschreven :";K$
2240 IF LEN(K$)>6 THEN K$=LEFT$(K$,6)
2250 IF Q$="I" THEN 2280
2260 IF Q$="D" THEN 2310
2270 GOTO 2160
2280 MOTOR ON
2290 OPEN"CAS:K$" FOR OUTPUT AS#1
2300 GOTO 2320
2310 OPEN K$ FOR OUTPUT AS#1
2320 PRINT#1,C
2330 PRINT#1,N
2340 FOR Y=1 TO C
2350 PRINT#1,A$(Y)
2360 NEXT Y
2370 CLOSE#1
2380 IF Q$="I" THEN MOTOR OFF
2390 LOCATE 14,21:PRINT"[M]=MENU "
2400 Q$=INKEY$:IF Q$=""THEN 2400
2410 IF Q$="M" OR Q$="m" THEN RETURN ELS
E 2400
2420 REM *****
      ** LITERATUURLIJST INLEZEN**
      *****

```



```

2430 CLS
2440 PRINT"L I T E R A T U U R O N D E R
  Z O E K":PRINT"LITERATUURLIJST INLEZEN"
:PRINT STRING$(39,"-"):PRINT
2450 PRINT"WILT U INLEZEN VAN TAPE OF DI
SC ?"
2460 PRINT:INPUT"(T) of (D):";Q$
2470 PRINT:PRINT
2480 INPUT"Onder welke NAAM moet de FILE
  worden      INGELEZEN :";K$
2490 IF LEN(K$)>6 THEN K$=LEFT$(K$,6)
2500 IF Q$="T" THEN 2530
2510 IF Q$="D" THEN 2560
2520 GOTO 2420
2530 MOTOR ON
2540 OPEN"CAS:K$" FOR INPUT AS#1
2550 GOTO 2570
2560 OPEN K$ FOR INPUT AS#1
2570 INPUT#1,C
2580 INPUT#1,N
2590 FOR Y=1 TO C
2600 INPUT#1,A$(Y)
2610 NEXT Y
2620 CLOSE#1
2630 IF Q$="T" THEN MOTOR OFF
2640 LOCATE 14,21:PRINT"[M]=MENU "
2650 Q$=INKEY$:IF Q$=""THEN 2650
2660 IF Q$="M" OR Q$="m" THEN RETURN ELS
E 2650
2670 REM *****
      ** STOPPEN MET PROGRAMMA **
      *****

2680 CLS
2690 PRINT"L I T E R A T U U R O N D E R
  Z O E K":PRINT"STOPPEN MET PROGRAMMA":P
RINT STRING$(39,"-"):PRINT
2700 LOCATE 0,5:PRINT"U WILT STOPPEN MET
  HET PROGRAMMA ?"

```

```
2710 PRINT"Heeft U de gegevens die U hee  
ft inge- voerd al vastgelegd op Casset  
te of op Diskette ?"  
2720 PRINT:PRINT"Als dit nog niet is geb  
eurd TOETST U [M] voor terugkeer naar  
MENU, is het NIET NODIG de gegevens  
vast te leggen ofals dit al is gedaan ,  
Toetst U [E] om programma te beëindigen  
."  
2730 LOCATE 0,21:PRINT"[M] = MENU [E]=  
EINDIGEN met programma"  
2740 Q$=INKEY$:IF Q$=""THEN 2740  
2750 IF Q$="M" OR Q$="m" THEN RETURN  
2760 IF Q$="E" OR Q$="e" THEN END ELSE 2  
740
```

KLANTENINFORMATIE

Dit programma is bedoeld voor het kleine bedrijf, dat met de MSX op snelle en eenvoudige wijze de gegevens over de klanten bij de hand wilt hebben.

Met dit programma legt U de volgende gegevens over Uw klanten vast:

Naam
Adres
Postcode + Woonplaats
Telefoon
Contactpersoon
Soort Klant
Debiteurennummer
Opmerking

Het programma is ook weer volledig menugestuurd.

Vanuit het Hoofdmenu kunt U direct naar de diverse routines springen en datgene uitvoeren dat U wilde doen.

In het programma zijn een aantal soortcodes al ingevoerd. U moet die codes, die tevens dienen om de klanten te sorteren, uiteraard voor Uw eigen toepassing aanpassen.

U kunt net zoveel soortcodes gebruiken als U wenst. U dient daarvoor de volgende regels aan te passen: regels 2160-2250 en regel 290 van de invoerroutine.

In totaal kunt U per file, die U wegschrijft naar tape of disk, gegevens over 150 klanten opslaan.

Het gehele programma is voorzien van duidelijke REM-statements, zodat het programma eenvoudig te volgen is.

```

10 REM *****
    **      CLIE@NTIEN-INFORMATIE      **
    **      voor MSX computers          **
    ** (c)1985 Robert-Jan Donkers     **
    *****

20 SCREEN 0:WIDTH 40:COLOR 15,1,1:KEYOFF
30 CLEAR 15000
40 DIM S$(150),N$(150),A$(150),P$(150),W
   $(150),D$(150),T$(150),C$(150),O$(150)
50 Z$="Clie@nten-Informatie (c) 1985
   TERMINAL"
60 REM *****
    *S=SOORT N=NAAM A=ADRES P=P.Code*
    *W=WOONPL. D=DEB.NR. T=TELEFOON *
    *C=CONTACTPERS. O=OPMERKING *
    *****

70 A=0:B=0:GOTO 2450
80 POKE 64683!,0:CLS:PRINT Z$:PRINT STRI
   NG$(39,"-"):PRINT
90 LOCATE 15,4:PRINT"M E N U":PRINT SPC(
   15);"-----"
100 LOCATE 5,7:PRINT"[1].....Gegevens i
   nvoeren"
110 LOCATE 5,9:PRINT"[2].....Gegevens v
   eranderen"
120 LOCATE 5,11:PRINT"[3].....Gegevens
   opzoeken"
130 LOCATE 5,13:PRINT"[4].....Gegevens
   opslaan"
140 LOCATE 5,15:PRINT"[5].....Gegevens
   inlezen"
150 LOCATE 5,17:PRINT"[6].....Printen a
   dresstroken"
160 LOCATE 5,19:PRINT"[7].....Programma
   be indigen"
170 LOCATE 5,22:PRINT"TOETS UW KEUZE IN
   [1] t/m [7]"
180 Q$=INKEY$:IF Q$="" THEN 180
190 IF Q$<"1" OR Q$>"7" THEN 180
200 Q=VAL(Q$)

```



```

210 ON Q GOTO 220,720,1110,1480,1800,212
0,2360
220 REM *****
**      GEGEVENS INVOEREN      **
*****

230 CLS:A=A+1:IF A=149 THEN PRINT" NA DI
T ADRES GEHEUGEN VOL !!":FOR Z=1 TO 1500
:NEXT Z
240 LOCATE 0,0:PRINT Z$:PRINT STRING$(39
,"-")
250 LOCATE 0,3:PRINT"Naam   : "
260 LOCATE 0,4:PRINT"Adres  : "
270 LOCATE 0,5:PRINT"P.Code: "
280 LOCATE 0,6:PRINT"Plaats: "
290 LOCATE 0,7:PRINT"Soort [B,C,G,P,D] :
"

300 PRINT:PRINT STRING$(39,"-")
310 LOCATE 0,11:PRINT"Deb.No. : "
320 LOCATE 0,13:PRINT"Telefoon: "
330 LOCATE 0,15:PRINT"Contact  : "
340 LOCATE 0,17:PRINT"Opmerking: "
350 LOCATE 0,18:PRINT STRING$(9,"-")
360 LOCATE 0,11:PRINT"Deb.No. : "
370 LOCATE 0,13:PRINT"Telefoon: "
380 LOCATE 0,15:PRINT"Contact  : "
390 LOCATE 0,17:PRINT"Opmerking: "
400 LOCATE 0,18:PRINT STRING$(9,"-")
410 LOCATE 9,3:INPUT K$
420 IF LEN(K$)>24 THEN K$=LEFT$(K$,24)
430 N$(A)=K$
440 LOCATE 9,4:INPUT K$
450 IF LEN(K$)>24 THEN K$=LEFT$(K$,24)
460 A$(A)=K$
470 LOCATE 9,5:INPUT K$
480 IF LEN(K$)>7 THEN 470
490 P$(A)=K$
500 LOCATE 9,6:INPUT K$
510 IF LEN(K$)>14 THEN K$=LEFT$(K$,14)
520 W$(A)=K$
530 LOCATE 20,7:INPUT K$

```

```

540 IF LEN(K$)>1 THEN 530
550 S$(A)=K$
560 LOCATE 10,11:INPUT K$
570 IF LEN(K$)>10 THEN 560
580 D$(A)=K$
590 LOCATE 10,13:INPUT K$
600 IF LEN(K$)>12 THEN 590
610 T$(A)=K$
620 LOCATE 10,15:INPUT K$
630 IF LEN(K$)>24 THEN 620
640 C$(A)=K$
650 LOCATE 0,19:INPUT K$
660 IF LEN(K$)>80 THEN 650
670 O$(A)=K$
680 LOCATE 0,22:PRINT"      [M]=MENU      [C
]=Volgende KLANT"
690 Q$=INKEY$:IF Q$="" THEN 690
700 IF Q$="C" OR Q$="c" THEN 220
710 IF Q$="M" OR Q$="m" THEN 80 ELSE 690
720 REM *****
**      GEGEVENS VERANDEREN      **
*****
730 CLS:PRINT Z$:PRINT STRING$(39,"-")
740 PRINT"VERANDEREN VAN GEGEVENS":PRINT
750 IF A=0 THEN LOCATE 10,10:PRINT"BESTA
ND IS LEEG !!!":FOR Y=1 TO 1000:NEXT Y:G
OTO 80
760 INPUT"NAAM: ";X$
770 IF LEN(X$)>24 THEN X$=LEFT$(X$,24)
780 INPUT"PLAATS: ";Y$
790 IF LEN(Y$)>14 THEN Y$=LEFT$(Y$,14)
800 B=1
810 FOR K=1 TO 150
820 IF N$(K)=X$ AND W$(K)=Y$ THEN GOSUB
2480:GOTO 840
830 NEXT K:B=0:GOTO 1290
840 LOCATE 0,22:INPUT"Welk NUMMER wilt U
veranderen ? ";Q
850 IF Q<1 OR Q>9 THEN 840
860 LOCATE 0,22:PRINT STRING$(39," ")

```

```

870 IF Q=1 THEN LOCATE 0,3:PRINT CHR$(23
6):LOCATE 2,3:INPUT K$:IF LEN(K$)>24 THE
N K$=LEFT$(K$,24)
880 IF Q=1 THEN N$(K)=K$
890 IF Q=2 THEN LOCATE 0,4:PRINT CHR$(23
6):LOCATE 2,4:INPUT K$:IF LEN(K$)>24 THE
N K$=LEFT$(K$,24)
900 IF Q=2 THEN A$(K)=K$
910 IF Q=3 THEN LOCATE 0,5:PRINT CHR$(23
6):LOCATE 2,5:INPUT K$:IF LEN(K$)>7 THEN
910
920 IF Q=3 THEN P$(K)=K$
930 IF Q=4 THEN LOCATE 0,6:PRINT CHR$(23
6):LOCATE 2,6:INPUT K$:IF LEN(K$)>14 THE
N K$=LEFT$(K$,14)
940 IF Q=4 THEN W$(K)=K$
950 IF Q=5 THEN LOCATE 0,7:PRINT CHR$(23
6):LOCATE 2,7:INPUT K$:IF LEN(K$)>1 THEN
950
960 IF Q=5 THEN S$(K)=K$
970 IF Q=6 THEN LOCATE 0,11:PRINT CHR$(2
36):LOCATE 2,11:INPUT K$:IF LEN(K$)>10 T
HEN 970
980 IF Q=6 THEN D$(K)=K$
990 IF Q=7 THEN LOCATE 0,13:PRINT CHR$(2
36):LOCATE 2,13:INPUT K$:IF LEN(K$)>12 T
HEN K$=LEFT$(K$,12)
1000 IF Q=7 THEN I$(K)=K$
1010 IF Q=8 THEN LOCATE 0,15:PRINT CHR$(
236):LOCATE 2,15:INPUT K$:IF LEN(K$)>24
THEN K$=LEFT$(K$,24)
1020 IF Q=8 THEN C$(K)=K$
1030 IF Q=9 THEN LOCATE 0,19:PRINT CHR$(
236):LOCATE 2,19:INPUT K$:IF LEN(K$)>80
THEN K$=LEFT$(K$,80)
1040 IF Q=9 THEN O$(K)=K$
1050 LOCATE 0,22:PRINT"Nog meer verander
en deze klant [J]/[N]"
1060 Q$=INKEY$:IF Q$="" THEN 1060
1070 IF Q$="J" OR Q$="j" THEN 840

```



```

1080 IF Q$="N" OR Q$="n" THEN LOCATE 0,2
2:PRINT"          Toets [RETURN] voor ME
NU          ":ELSE 1060
1090 Q$=INKEY$: IF Q$="" THEN 1090
1100 IF ASC(Q$)-13 THEN 80 ELSE 1090
1110 REM *****
      **   GEGEVENS OPZOEKEN   **
      *****
1120 B=0:CLS:PRINT 2$:PRINT STRING$(39,"
"):PRINT
1130 IF A=0 THEN LOCATE 10,10:PRINT"BEST
AND IS LEEG !!!":FOR Y=1 TO 1000:NEXT Y:
GOTO 80
1140 PRINT"ZOEKMOGELIJKHEDEN:"
1150 PRINT SPC(10);"[1]... Op NAAM"
1160 PRINT SPC(10);"[2]... Op PLAATS"
1170 PRINT SPC(10);"[3]... Op DEB.NO."
1180 PRINT SPC(10);"[4]... Alle Klanten"
1190 LOCATE 0,22:PRINT" TOETS Uw KEUZE
IN [1] t/m [4]"
1200 Q$=INKEY$: IF Q$="" THEN 1200
1210 IF Q$<"1" OR Q$>"4" THEN 1200
1220 Q=VAL(Q$)
1230 ON Q GOTO 1240,1320,1380,1440
1240 LOCATE 0,12:INPUT"NAAM:";Y$
1250 IF LEN(Y$)>24 THEN Y$=LEFT$(Y$,24)
1260 FOR K=1 TO 150
1270 IF N$(K)=Y$ THEN GOSUB 2480
1280 NEXT K
1290 CLS:LOCATE 0,22:PRINT" EINDE van o
verzicht !! TOETS [RETURN]"
1300 Q$=INKEY$: IF Q$="" THEN 1300
1310 IF ASC(Q$)=13 THEN 80 ELSE 1300
1320 LOCATE 0,12:INPUT"PLAATS:";Y$
1330 IF LEN(Y$)>24 THEN Y$=LEFT$(Y$,24)
1340 FOR K=1 TO 150
1350 IF W$(K)=Y$ THEN GOSUB 2480
1360 NEXT K
1370 GOTO 1290
1380 LOCATE 0,12:INPUT"DEB.NO.:";Y$

```



```

1390 IF LEN(Y$)>10 THEN Y$=LEFT$(Y$,10)
1400 FOR K=1 TO 150
1410 IF D$(K)=Y$ THEN GOSUB 2480
1420 NEXT K
1430 GOTO 1290
1440 CLS:FOR K=1 TO 150
1450 GOSUB 2480
1460 NEXT K
1470 GOTO 1290
1480 REM *****
      **   GEGEVENS   OPSLAAN   **
      *****
1490 POKE 64683!,255:CLS:PRINT Z$:PRINT
STRING$(39,"-"):PRINT"GEGEVENS OPSLAAN":
PRINT
1500 IF A=0 THEN PRINT"BESTAND IS LEEG !
!!":GOTO 1770
1510 PRINT"Wilt U wegschrijven naar TAPE
of DISC ?"
1520 PRINT:INPUT"[I] of [D] :";Q$
1530 PRINT:PRINT
1540 INPUT"Onder welke NAAM moet de FILE
worden   weggeschreven :";K$
1550 IF LEN(K$)>6 THEN K$=LEFT$(K$,6)
1560 IF Q$="I" THEN 1590
1570 IF Q$="D" THEN 1620
1580 GOTO 1480
1590 MOTOR ON
1600 OPEN "CAS:K$" FOR OUTPUT AS#1
1610 GOTO 1630
1620 OPEN K$ FOR OUTPUT AS#1
1630 PRINT #1,A
1640 FOR Y=1 TO A
1650 PRINT #1,S$(Y)
1660 PRINT #1,N$(Y)
1670 PRINT #1,A$(Y)
1680 PRINT #1,P$(Y)
1690 PRINT #1,W$(Y)
1700 PRINT #1,D$(Y)
1710 PRINT #1,T$(Y)

```

```

1720 PRINT #1,C$(Y)
1730 PRINT #1,O$(Y)
1740 NEXT Y
1750 CLOSE#1
1760 IF Q$="T" THEN MOTOR OFF
1770 LOCATE 14,22:PRINT"Toets [RETURN]"
1780 Q$=INKEY$:IF Q$="" THEN 1780
1790 IF ASC(Q$)=13 THEN 80 ELSE 1770
1800 REM *****
      **   GEGEVENS INLEZEN   **
      *****
1810 POKE 64683!,255:CLS:PRINT Z$:PRINT
STRING$(39,"-"):PRINT"GEGEVENS INLEZEN":
PRINT
1820 IF A=0 THEN PRINT"BESTAND IS LEEG !
!!":GOTO 2090
1830 PRINT"Wilt U inlezen van TAPE of DI
SC ?"
1840 PRINT:INPUT"[T] of [D] :";Q$
1850 PRINT:PRINT
1860 INPUT"Onder welke NAAM moet de FILE
worden   ingelezen :";K$
1870 IF LEN(K$)>6 THEN K$=LEFT$(K$,6)
1880 IF Q$="T" THEN 1910
1890 IF Q$="D" THEN 1940
1900 GOTO 1800
1910 MOTOR ON
1920 OPEN "CAS:K$" FOR INPUT AS#1
1930 GOTO 1950
1940 OPEN K$ FOR INPUT AS#1
1950 INPUT #1,A
1960 FOR Y=1 TO A
1970 INPUT #1,S$(Y)
1980 INPUT #1,N$(Y)
1990 INPUT #1,A$(Y)
2000 INPUT #1,P$(Y)
2010 INPUT #1,W$(Y)
2020 INPUT #1,D$(Y)
2030 INPUT #1,T$(Y)
2040 INPUT #1,C$(Y)

```

```

2050 INPUT #1,0$(Y)
2060 NEXT Y
2070 CLOSE#1
2080 IF Q$="I" THEN MOTOR OFF
2090 LOCATE 14,22:PRINT"Toets [RETURN]"
2100 Q$=INKEY$:IF Q$="" THEN 2100
2110 IF ASC(Q$)=13 THEN 80 ELSE 2090
2120 REM *****
      ** ADRESSTROKEN PRINTEN **
      *****
2130 CLS:POKE 64683!,255:PRINT Z$:PRINT
STRING$(39,"-"):PRINT"ADRESSTROKEN PRINT
EN":PRINT
2140 IF A=0 THEN LOCATE 10,10:PRINT"BEST
AND IS LEEG !!!":FOR Y=1 TO 1000:NEXT Y:
GOTO 80
2150 PRINT"Geef allereerst de KEUZE aan:
":PRINT
2160 PRINT"[B].....Boekhandels"
2170 PRINT"[C].....Computerwinkels"
2180 PRINT"[G].....Grossiers"
2190 PRINT"[P].....Partikulieren"
2200 PRINT"[D].....Diverse Klanten"
2210 PRINT"[A].....Alle Klanten"
2220 LOCATE 0,22:PRINT"Toets Keuze in [B
],[C],[G],[P],[D],[A]"
2230 Q$=INKEY$:IF Q$="" THEN 2230
2240 IF Q$="B" OR Q$="C" OR Q$="A" OR Q$
="D" OR Q$="G" OR Q$="P" THEN 2250 ELSE
2230
2250 Y$=Q$:IF Y$="A" THEN 2310
2260 FOR K=1 TO 150
2270 IF S$(K)=Y$ THEN LPRINT:LPRINT N$(K
):LPRINT A$(K):LPRINT P$(K);" ";W$(K):
LPRINT:LPRINT STRING$(40,"-")
2280 NEXT K
2290 LOCATE 0,22:PRINT"          Toets [RE
TURN] voor MENU          "
2295 Q$=INKEY$:IF Q$="" THEN 2295
2300 IF ASC(Q$)=13 THEN 80 ELSE 2295
2310 FOR K=1 TO 150

```

```

2320 IF N$(K)=" " THEN 2290
2330 LPRINT:LPRINT N$(K):LPRINT A$(K):LP
RINT P$(K); " ";W$(K):LPRINT:LPRINT STR
ING$(40,"-")
2340 NEXT K
2350 GOTO 2290
2360 REM *****
      ** PROGRAMMA BEEINDIGEN **
      *****
2370 CLS:PRINT Z$:PRINT STRING$(39,"-"):
PRINT"STOPPEN MET PROGRAMMA":PRINT
2380 LOCATE 0,5:PRINT"U wilt stoppen met
het programma ?"
2390 PRINT"Heeft U de gegevens die U hee
ft inge- voerd al vastgelegd op Casset
te of Diskette ?"
2400 PRINT:PRINT"Als dit nog niet is geb
eurd TOETST U [M] voor terugkeer naar
MENU. Is het NIET NODIG de ge
gevens vast te leggen of als dit al ge
daan is, TOETST U [E] om het programma
te beëindigen."
2410 LOCATE 0,22:PRINT" [M]=MENU [E]=
EINDIGEN met programma"
2420 Q$=INKEY$: IF Q$=" " THEN 2420
2430 IF Q$="M" OR Q$="m" THEN 80
2440 IF Q$="E" OR Q$="e" THEN END ELSE 2
420
2450 CLS:POKE 64683!,255:LOCATE 5,10:INP
UT"Uw Code No: ";Y$
2460 IF Y$="5110ACTSP" THEN 80
2470 LOCATE 5,20:PRINT"GEEN TOEGANG":FOR
K=1 TO 500:NEXT K:END
2480 REM *****
      * ALGEMENE SUBROUTINE VOOR: *
      * ZOEKEN EN VERANDEREN *
      *****
2490 CLS:PRINT Z$:PRINT STRING$(39,"-")
2500 LOCATE 0,3:PRINT "1: ";N$(K)
2510 LOCATE 0,4:PRINT "2: ";A$(K)

```



```
2520 LOCATE 0,5:PRINT "3:";P$(K)
2530 LOCATE 0,6:PRINT "4:";W$(K)
2540 LOCATE 0,7:PRINT "5:";S$(K)
2550 LOCATE 0,11:PRINT "6:";D$(K)
2560 LOCATE 0,13:PRINT "7:";T$(K)
2570 LOCATE 0,15:PRINT "8:";C$(K)
2580 LOCATE 0,19:PRINT "9:";O$(K)
2590 IF B=1 THEN RETURN
2600 LOCATE 0,22:PRINT" [RETURN]=VERVOLG
      [M]=MENU"
2610 Q$=INKEY$:IF Q$="" THEN 2610
2620 IF Q$="" OR Q$="m" THEN 80
2630 IF ASC(Q$)=13 THEN RETURN ELSE 2610
```

TELEFOONINDEX

Dit programma is niet zo maar een telefoonnummerbestand er zijn een aantal aardigheidjes van MSX-BASIC ingebouwd die U ook in Uw eigen programma's zou kunnen toepassen.

Zo is er gebruik gemaakt van een aantal POKE-instructies die U overal zou kunnen gebruiken.

o.a.: POKE 64683,255 voor CAPS LOCK
 POKE 62387,40 zet 40 karakters op
 een regel.
 POKE 62385,23 zet het scherm op 23
 regels, de bovenste
 regel wordt dan niet
 meer gebruikt.

Op de bovenste schermregel zetten we in kleine letters 'telefoonnummers bestand'.

Door direct daarna POKE 62385,23 uit te voeren is deze regel niet meer te wissen door CLS.

Daarna worden door de UPOKE opdrachten in regels 200-210 de kleine letters en de cijfers veranderd in INVERSE VIDEO karakters.

Daarom zijn er in de diverse MENU's en Opdrachtregels kleine letters gebruikt. U moet de teksten met kleine letters zo 'letterlijk' mogelijk overnemen, anders schopt U het programma door elkaar.

Het programma houdt alleen de naam + telefoonnummer bij.

Gebruik in de naam NOOIT een komma om onderscheidt te maken tussen gelijke achternamen, maar gebruik dan een PUNT. , anders krijgt U de foutmelding: 'EXTRA IGNORED'.

Gebruik dus NOOIT: Jansen,H
Jansen,P

MAAR GEBRUIK: Jansen.H
Jansen.P

Zet ook altijd eerst de achternaam en dan de voorletters.

In alle bestandsprogramma's in dit boek is het aan te bevelen in namen,adressen,titels etc. GEEN KOMMA's maar PUNTEN te gebruiken. U kunt in met dit programma zo'n 400 telefoonnummers opslaan.

Zoeken kunt U op nummer of sleutelwoord. Het voordeel bij het zoeken is dat U beslist niet het volledige nummer hoeft in te voeren. Heeft U bijv. een kennis in VENLO, dan is voldoende om het netnummer '077' in te toetsen en U krijgt op het scherm alle nummers van VENLO uit Uw bestand op het scherm. Toetst U alleen '07' in dan krijgt U alle nummers die beginnen met '07', zoals de netnummers 071,073,075,076,077 etc.

Zoeken op sleutelwoord werkt precies hetzelfde. U kunt een volledige naam invoeren maar ook volstaan met een aantal beginletters bijv. 'JAN'.

U vindt dan bijv.

JAN	099-432156
JANS	078-321564
JANSEN.H	634521
JANSEN.P	035-440044
JANSSEN	020-8246517
etc.etc.	

Toetst U alleen 'J' in dan komen alle namen op het scherm die beginnen met een 'J'. U ziet eenvoudiger zoeken kan bijna niet. Het programma is voorzien van duidelijke REM-statements, zodat U het programma goed kunt volgen.

```

10 COLOR 15,4,4
20 REM *****
   ** TELEFOONNUMMERS BESTAND **
   *****
30 REM *****
   **      INITIALISEREN      **
   *****

40 CLS
50 CLEAR 16000
60 TAL=400:REM MAX. AANTAL NUMMERS
70 WIDTH 40
80 KEYOFF
90 COLOR 1,1,1
100 AA=0
110 POKE 64683!,255
120 DIM N$(TAL),I$(TAL),NI(TAL)
130 PRINT "***** telefoonnummers best
and *****"
140 POKE 62387!,40
150 SCREEN 0
160 POKE 62385!,23
170 FOR F=1 TO 400
180 N$(F)="#":NI(F)=0
190 NEXT F
200 FOR F=776 TO 983:VPOKE BASE(2)+F,255
-VPEEK(BASE(2)+F-256):NEXT F
210 FOR F=384 TO 463:VPOKE BASE(2)+F,255
-VPEEK(BASE(2)+F):NEXT F
220 COLOR 11,1,1
230 REM *****
   **      HOOFD-MENU      **
   *****

240 CLS
250 LOCATE 17,2:PRINT "menu"
260 PRINT:PRINT:PRINT TAB(4);"nummers op
zoeken-----=> 1"
270 PRINT:PRINT TAB(4);"zoeken met sleut
elwoorden---=> 2"
280 PRINT:PRINT TAB(4);"alle nummers pri
nten-----=> 3"

```



```

290 PRINT:PRINT TAB(4);"nummers invoeren
-----> 4"
300 PRINT:PRINT TAB(4);"nummers afvoeren
-----> 5"
310 PRINT:PRINT TAB(4);"bestand laden---
-----> 6"
320 PRINT:PRINT TAB(4);"bestand save---
-----> 7"
330 PRINT:PRINT:PRINT TAB(11);"toets uw
keuze in"
340 A$=INKEY$:IF A$="" THEN 340
350 IF A$<"1" OR A$>"7" THEN 340
360 Q=VAL(A$)
370 CLS
380 ON Q GOTO 390,480,700,840,1100,1320,
1400
390 REM *****
      ** ZOEKEN VAN EEN NUMMER **
      *****

395 W=0
400 IF AA<1 THEN LOCATE 10,10:PRINT "het
bestand is leeg":FOR F=1 TO 1000:NEXT F
:GOTO 230
410 LOCATE 3,6:S$="":INPUT "naam ";S$
420 IF LEN(S$)<1 OR LEN(S$)>25 THEN 410
430 FOR F=1 TO AA
440 IF S$=LEFT$(N$(F),LEN(S$)) THEN PRIN
T:PRINT N$(F),"tel:";T$(F):W=1
450 NEXT F
460 IF W=0 THEN LOCATE 3,8:PRINT "deze n
aam zit niet in het bestand"
470 GOTO 1600
480 REM *****
      ** ZOEKEN MET SLEUTELWOORD **
      *****

490 IF AA<1 THEN LOCATE 10,10:PRINT "het
bestand is leeg":FOR F=1 TO 1000:NEXT F
:GOTO 230
500 LOCATE 4,4:PRINT"WELKE SLEUTEL WILT
U GEBRUIKEN ?"

```

```

510 PRINT:PRINT TAB(8);"nAAM OF tELEFOON
NUMMER ?"
520 A$=INKEY$:IF A$="" THEN 520
530 IF A$="N" THEN CLS:GOTO 630
540 IF A$<>"T" THEN 510
550 CLS
560 LOCATE 0,2:SS$="":INPUT "WAT IS DE ST
EUTEL ";SS
570 IF LEN(SS$)<1 OR LEN(SS$)>25 THEN 560
580 PT=0
590 FOR F=1 TO AA
600 IF SS$=LEFT$(I$(F),LEN(SS$)) THEN PRIN
T N$(F);TAB(25);I$(F):PRINT:PT=PT+1
604 IF PT>9 THEN IF F<AA THEN PRINT:PRIN
T TAB(7);"return = VOLGENDE PAGINA.":GOT
O 608
606 GOTO 610
608 IF INKEY$<>CHR$(13) THEN 608 ELSE PT
=0:CLS
610 NEXT F
620 GOTO 1600
630 LOCATE 0,2:SS$="":INPUT "WAT IS DE SL
EUTEL ";SS
640 IF LEN(SS$)<1 OR LEN(SS$)>14 THEN 630
645 PT=0
650 FOR F=1 TO AA
660 IF SS$=LEFT$(N$(F),LEN(SS$)) THEN PRIN
T:PRINT N$(F);TAB(25);I$(F):PT=PT+1
664 IF PT>9 THEN IF F<AA THEN PRINT:PRIN
T TAB(7);"return = VOLGENDE PAGINA.":GOT
O 668
666 GOTO 670
668 IF INKEY$<>CHR$(13) THEN 668 ELSE PT
=0:CLS
670 NEXT F
680 PRINT:PRINT
690 GOTO 1600
700 REM *****
      ** ALLE NUMMERS PRINTEN **
      *****

```

```

710 IF AA<1 THEN LOCATE 10,10:PRINT "het
bestand is leeg":FOR F=1 TO 1000:NEXT F
:GOTO 230
720 PT=0
730 FOR F=1 TO AA
740 PRINT N$(F);TAB(25);T$(F)
750 PRINT
760 PT=PT+1
770 IF PT>9 THEN IF F<AA THEN PRINT:PRIN
T TAB(7);"return = VOLGENDE PAGINA.":GOT
O 790
780 GOTO 800
790 IF INKEY$<>CHR$(13) THEN 790 ELSE PT
=0:CLS
800 NEXT F
810 PRINT:PRINT
820 PT=0
830 GOTO 1600
840 REM *****
** NUMMERS INVOEREN **
*****
850 LOCATE 16,2:PRINT "invoeren"
860 IF AA+1>TAL THEN LOCATE 11,9:PRINT "
het bestand is vol":FOR F=1 TO 1000:NEXT
F:GOTO 230
870 LOCATE 0,5:INPUT "naam ";S$
880 IF LEN(S$)<1 OR LEN(S$)>25 THEN 870
890 LOCATE 0,7:INPUT "telefoonnummer ";D
$
900 IF LEN(D$)>15 THEN 890
910 LOCATE 8,21:PRINT "GEGEVENS IN ORDE
? (j/n)"
920 A$=INKEY$:IF A$="" THEN 920
930 IF A$="N" THEN CLS:GOTO 870
940 IF A$<>"J" THEN 920
950 AA=AA+1:N$(AA)=S$:T$(AA)=D$
960 LL=LEN(N$(AA)):IF LL<8 THEN N$(AA)=N
$(AA)+SPACE$(8-LL)
970 XX=8
980 TI=100^XX

```

```

990 FOR F=1 TO XX
1000 IF MID$(N$(AA),F,1)=CHR$(32) THEN 1
040
1010 NI(AA)=NI(AA)+(IT*(ASC(MID$(N$(AA),
F,1))-64))
1020 IT=IT/100
1030 NEXT F
1040 IF AA<2 THEN 1090
1050 FOR F=AA TO 2 STEP -1
1060 IF NI(F)>=NI(F-1) THEN 1090
1070 SWAP N$(F),N$(F-1):SWAP I$(F),I$(F-
1):SWAP NI(F),NI(F-1)
1080 NEXT F
1090 GOTO 1600
1100 REM*****
      **      NUMMERS AFVOEREN      **
      *****
1110 IF AA<1 THEN LOCATE 10,10:PRINT "he
t bestand is leeg":FOR F=1 TO 1000:NEXT
F:GOTO 230
1120 LOCATE 16,2:PRINT "afvoeren"
1130 LOCATE 0,7
1140 INPUT "naam ";S$
1150 IF LEN(S$)<1 OR LEN(S$)>25 THEN 113
0
1160 FOR F=1 TO AA
1170 IF S$=LEFT$(N$(F),LEN(S$)) THEN 120
0
1180 NEXT F
1190 PRINT:PRINT TAB(3);"deze naam zit n
iet in het bestand":GOTO 1600
1200 PRINT:PRINT:PRINT "naam";TAB(25);"n
ummer"
1210 PRINT:PRINT N$(F);TAB(25);I$(F)
1220 LOCATE 12,21:PRINT"AFVOEREN ? (j/n)
"
1230 A$=INKEY$: IF A$="" THEN 1230
1240 IF A$="N" THEN 1600
1250 IF A$<>"J" THEN 1230
1260 N$(F)="#":I$(F)="":NI(F)=0
1270 FOR I=F TO AA

```



```

1280 SWAP N$(I),N$(I+1):SWAP I$(I),I$(I+
1):SWAP NI(I),NI(I+1)
1290 NEXT I
1300 AA=AA-1
1310 GOTO 1600
1320 REM*****
      **      BESTAND LADEN      **
      *****
1325 LOCATE 11,10:PRINT "EVEN GEDULD A.U
.B."
1330 OPEN "CAS:TELE" FOR INPUT AS 1
1340 INPUT #1,AA
1345 PRINT:PRINT TAB(5);"ER WORDEN";AA;"
NUMMERS GELADEN"
1350 FOR F=1 TO AA
1360 INPUT #1,N$(F),I$(F),NI(F)
1370 NEXT F
1380 CLOSE #1
1390 GOTO 230
1400 REM*****
      **      BESTAND SAVEN      **
      *****
1405 IF AA<1 THEN LOCATE 10,10:PRINT "he
t bestand is leeg":FOR F=1 TO 1000:NEXT
F:GOTO 230
1410 LOCATE 3,7:PRINT "MET WELKE BAUDRAI
E WILT U SAVEN ?"
1420 LOCATE 13,10:PRINT "1 = 1200 BAUD"
1430 PRINT TAB(13);"2 = 2400 BAUD"
1440 A$=INKEY$:IF A$="" THEN 1440
1450 IF A$<"1" OR A$>"2" THEN 1440
1460 BR=VAL(A$)
1470 SCREEN , , , BR
1480 CLS
1490 LOCATE 11,10:PRINT "EVEN GEDULD A.U
.B."
1510 OPEN "CAS:TELE" FOR OUTPUT AS 1
1520 PRINT #1,AA
1530 FOR F=1 TO AA
1540 PRINT #1,N$(F)

```

```
1550 PRINT #1,TS(F)
1560 PRINT #1,NI(F)
1570 NEXT F
1580 CLOSE #1
1590 GOTO 230
1600 LOCATE 8,21:PRINT "TERUG NAAR MENU
? (j/n) "
1610 AS=INKEY$:IF AS="-" THEN 1610
1620 IF AS="J" THEN 230
1630 IF AS<>"N" THEN 1610
1640 GOTO 370
```

BURO AGENDA

Zoals de titel van dit programma reeds doet vermoeden, is dit de gecomputeriseerde vervanger van de bekende kantooragenda.

U stuurt via het hoofdmenu het programma.

Er zijn de volgende mogelijkheden:

1. Notitie maken
2. Dag uit agenda op het scherm
3. Agenda saven
4. Agenda inlezen
5. Dag uit agenda wissen
6. Stoppen

Als U begint met het programma, wordt U eerst naar de datum van vandaag gevraagd.

Nu komt U in het hoofdmenu en kunt U notities maken in de agenda.

Allereerst geeft U de datum van de dag, waarvoor de notities gemaakt moeten worden.

Zit die datum al in de agenda, dan verschijnt de agenda voor die dag op het scherm en kunt U er notities bijmaken of veranderen.

Daarnaast heeft U in dit programma ook nog de mogelijkheid om een bepaalde dag te schrappen.

U kunt kris kras door elkaar verschillende datums invoeren, het programma zorgt er zelf voor dat alle datums keurig op volgorde komen te staan, zodat U bij het doorbladeren van de agenda een keurig overzicht krijgt van alle afspraken en notities.

Heeft U een nieuwe agenda gemaakt en U wilt die opslaan op cassette of diskette dan moet U eerst de oude agenda van de cassette resp. diskette verwijderen omdat de nieuwe agenda met dezelfde naam wordt weggeschreven. Bij een cassette kunt U gewoon de band

terugspoelen en de nieuwe agenda over de andere heen opnemen bij diskette moet de agenda verwijderd worden met: KILL "agenda". Als U met diskettes werkt moet U een aanpas- in het programma doen en wel de volgende:

Voeg de volgende regel toe:
1975 KILL "agenda": FOR Q=1 TO 500:NEXT Q

Door deze regel zal eerst de oude agenda van de schijf verwijderd worden en daarna wordt de nieuwe agenda weer op schijf gezet.

Na elke opdracht keert U weer terug naar het hoofdmenu door toets 'M' in te drukken of dat gebeurt automatisch zoals na het wegschrijven en inlezen.

Het programma is van duidelijke REM-statements voorzien, zodat de opbouw van het programma U snel duidelijk zal zijn.


```

10 REM *****
    **      BURO-AGENDA V1.0      **
    ** (C)1985 Robert-Jan Donkers**
    *****

20 SCREEN 0:WIDTH 40:COLOR 15,1,1:KEYOFF
30 CLEAR 15000
40 N=60:REM aantal dagen voor agenda
50 DIM A$(N),D$(18),B$(1)
60 Y$=SPACE$(39):C=0
70 Z$=SPACE$(34)
80 CLS:INPUT"geef de datum van vandaag:
      (dd-mm-jjjj) ";F$
90 IF LEN(F$)<>10 THEN 80
100 REM *****
    **      H O O F D M E N U      **
    *****

110 CLS:PRINT"BURO-AGENDA":PRINT STRING$(
39,"-"):PRINT:PRINT STRING$(39,"-")
120 LOCATE 25,0:PRINT">>> ";F$
130 LOCATE 30,2:PRINT"HOOFDMENU"
140 W=0
150 LOCATE 5,7:PRINT"[1].....Notities
maken"
160 LOCATE 5,9:PRINT"[2].....Dag uit a
genda op scherm"
170 LOCATE 5,11:PRINT"[3].....Agenda S
aven"
180 LOCATE 5,13:PRINT"[4].....Agenda I
nlezen"
190 LOCATE 5,15:PRINT"[5].....Dag uit
agenda wissen"
200 LOCATE 5,17:PRINT"[6].....Stoppen
met programma"
210 LOCATE 5,22:PRINT"Toets Uw keuze in
[1] t/m [6]"
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$<"1" OR Q$>"6" THEN 220
240 Q=VAL(Q$)
250 ON Q GOTO 1330,360,1870,2080,1740,22
80

```

```

260 LOCATE 0,4:PRINT"0900:"
270 LOCATE 0,6:PRINT"1000:"
280 LOCATE 0,8:PRINT"1100:"
290 LOCATE 0,10:PRINT"1200:"
300 LOCATE 0,12:PRINT"1300:"
310 LOCATE 0,14:PRINT"1400:"
320 LOCATE 0,16:PRINT"1500:"
330 LOCATE 0,18:PRINT"1600:"
340 LOCATE 0,20:PRINT"N.B.:"
350 RETURN
360 REM *****
      ** Notities op het scherm **
      *****
370 CLS:PRINT"BURO-AGENDA":PRINT STRING$(
(39,"-"):PRINT:PRINT STRING$(39,"-")
380 LOCATE 19,2:PRINT"AGENDA OP HET SCHE
RM"
390 IF C=0 THEN LOCATE 5,8:PRINT"AGENDA
IS NOG LEEG !!!":FOR Y=1 TO 1000:NEXT Y:
GOTO 100
400 LOCATE 5,8:PRINT"[1]....AGENDA raad
plegen"
410 LOCATE 5,10:PRINT"[2]....Bepaalde D
AG raadplegen"
420 LOCATE 5,12:PRINT"[3]....Terug naar
Hoofdmenu"
430 LOCATE 5,22:PRINT"Toets Uw keuze in
[1] t/m [3]"
440 Q$=INKEY$:IF Q$="" THEN 440
450 IF Q$<"1" OR Q$>"3" THEN 440
460 Q=VAL(Q$)
470 ON Q GOTO 480,710,100
480 FOR Y=8 TO 22:LOCATE 0,Y:PRINT Y$:NE
XT Y
490 GOSUB 260
500 FOR K=1 TO C
510 E$=LEFT$(A$(K),10)
520 LOCATE 19,0:PRINT">>> ";E$
530 P=11
540 L=4

```

```

550 LOCATE 5,L:
560 Q=ASC(MID$(A$(K),P,1))
570 P=P+1
580 IF Q>127 THEN 610
590 PRINT CHR$(Q);
600 GOTO 560
610 PRINT CHR$(Q-128)
620 L=L+1:IF L=22 THEN 630 ELSE 550
630 LOCATE 5,22:PRINT"Toets [RETURN] voo
r volgende dag"
640 Q$=INKEY$:IF Q$="" THEN 640
650 IF ASC(Q$)=13 THEN 660 ELSE 640
660 FOR Y=4 TO 22
670 LOCATE 5,Y:PRINT Z$
680 NEXT Y
690 NEXT K
700 GOTO 100
710 FOR Y=8 TO 22:LOCATE 0,Y:PRINT Y$:NE
XT Y
720 LOCATE 0,8:INPUT"Geef de gewenste da
tum          (dd-mm-jjjj):";G$
730 IF LEN(G$)<>10 THEN 720
740 LOCATE 0,8:PRINT Y$:PRINT Y$
750 FOR K=1 TO C
760 E$=LEFT$(A$(K),10)
770 IF E$=G$ THEN 800
780 NEXT K
790 GOTO 1300
800 GOSUB 260
810 B$=A$(K)
820 L=4
830 D=0:P=11
840 D=D+1:D$(D)=" "
850 LOCATE 5,L:
860 Q=ASC(MID$(B$,P,1))
870 P=P+1
880 IF Q>127 THEN 910
890 D$(D)=D$(D)+CHR$(Q):PRINT CHR$(Q);
900 GOTO 860
910 D$(D)=D$(D)+CHR$(Q):PRINTCHR$(Q-128)

```

```

920 L=L+1:IF L=22 THEN 930 ELSE 840
930 LOCATE 1,22:PRINT"Toets [M]=MENU
[N]=Nog 'n Notitie"
940 Q$=INKEY$:IF Q$="" THEN 940
950 IF Q$="m" OR Q$="M" THEN 100
960 IF Q$="N" OR Q$="n" THEN 970 ELSE 94
0
970 REM *****
** Toevoegen van 'n NOTITIE **
*****
980 LOCATE 0,22:PRINT" [RETURN]=Notiti
e blijft onveranderd "
990 L=4:B=5:P=11
1000 FOR D=1 TO 18
1010 I$="":J$=SPACE$(34)
1020 LOCATE B,L:PRINT CHR$(202)
1030 Q$=INKEY$:IF Q$="" THEN 1030
1040 IF ASC(Q$)=13 THEN 1090
1050 I$=I$+Q$
1060 LOCATE 5,L:PRINT I$
1070 B=B+1
1080 GOTO 1020
1090 I=LEN(I$)
1100 IF I=0 THEN 1180
1110 LOCATE B,L:PRINT" "
1120 C$=MID$(I$,I,1)
1130 MID$(J$,2,I+1)=I$
1140 MID$(I$,I,1)=CHR$(ASC(C$)+128)
1150 D$(D)=I$
1160 LOCATE 5,L:PRINT J$
1170 GOTO 1230
1180 LOCATE B,L:PRINT CHR$(32)
1190 E=LEN(D$(D))
1200 C$=MID$(D$(D),E,1)
1210 G$=MID$(D$(D),1,E-1)+CHR$(ASC(C$)-1
28)
1220 LOCATE 5,L:PRINT G$
1230 B=5:L=L+1
1240 NEXT D
1250 A$(K)=LEFT$(B$,10)

```



```

1260 FOR D=1 TO 18
1270 A$(K)=A$(K)+D$(D)
1280 NEXT D
1290 GOTO 100
1300 LOCATE 5,8:PRINT"Deze datum zit nie
t in de agenda !!"
1310 FOR Y=1 TO 1000:NEXT Y
1320 GOTO 100
1330 REM *****
      **          Notities maken          **
      *****
1340 CLS:PRINT"BURO-AGENDA":PRINT STRING
$(39,"-"):PRINT:PRINT STRING$(39,"-")
1350 LOCATE 20,2:PRINT"INVOER VAN NOTITI
ES"
1360 C=C+1
1370 LOCATE 5,8:INPUT"Datum: (dd-mm-jjjj)
:";Q$
1380 IF LEN(Q$)<>10 THEN 1370
1390 FOR Y= 1 TO C-1
1400 IF LEFT$(A$(Y),10)=Q$ THEN 1630
1410 NEXT Y
1420 A$(C)=A$(C)+Q$
1430 LOCATE 25,0:PRINT">>> ";Q$
1440 LOCATE 5,8:PRINT Y$
1450 GOSUB 260
1460 FOR L=4 TO 21
1470 LOCATE 5,L
1480 J$=SPACE$(34):I$=""
1490 INPUT I$
1500 IF I$="" THEN I$=" "
1510 K=LEN(I$)
1520 MID$(J$,2,K+1)=I$
1530 C$=MID$(I$,K,1)
1540 MID$(I$,K,1)=CHR$(ASC(C$)+128)
1550 A$(C)=A$(C)+I$
1560 LOCATE 5,L:PRINT J$
1570 NEXT L
1580 IF C>1 THEN GOSUB 1670

```

```

1590 LOCATE 5,22:PRINT"Toets [M]=Menu [
A]=Andere Dag"
1600 Q$=INKEY$:IF Q$="" THEN 1600
1610 IF Q$="M" OR Q$="m" THEN 100
1620 IF Q$="A" OR Q$="a" THEN 1340 ELSE
1600
1630 REM *****
      *deze dag zit al in de agenda*
      *****

1640 C=C-1:G$=Q$
1650 LOCATE 25,0:PRINT">>> ";Q$
1660 GOTO 740
1670 REM *****
      ** agenda op datum zetten **
      *****

1680 G$=LEFT$(A$(C),10)
1690 FOR Y=1 TO C
1700 C$=LEFT$(A$(Y),10)
1710 IF C$>G$ THEN SWAP A$(C),A$(Y):GOTO
1680
1720 NEXT Y
1730 RETURN
1740 REM *****
      * Dag uit agenda verwijderen *
      *****

1750 CLS:PRINT"BURO-AGENDA":PRINT STRING
$(39,"-"):PRINT:PRINT STRING$(39,"-")
1760 LOCATE 19,2:PRINT"DAG UIT AGENDA WI
SSEN"
1770 IF C=0 THEN LOCATE 5,8:PRINT"AGENDA
IS NOG LEEG !!!":FOR Y=1 TO 1000:NEXT Y
:GOTO 100
1780 LOCATE 0,8:INPUT"Geef de datum (dd-
mm-jjjj) ";G$
1790 IF LEN(G$)<>10 THEN 1750
1800 FOR Y=1 TO C
1810 C$=LEFT$(A$(Y),10)
1820 IF C$=G$ THEN A$(Y)="" :W=1
1830 IF A$(Y)="" THEN SWAP A$(Y),A$(Y+1)
1840 NEXT Y

```

```

1850 IF W=1 THEN C=C-1
1860 GOTO 100
1870 REM *****
      **          Agenda saven          **
      *****

1880 CLS:PRINT"BURO-AGENDA":PRINT STRING
$(39,"-"):PRINT:PRINT STRING$(39,"-")
1890 LOCATE 19,2:PRINT"AGENDA WEGSCHRIJ
EN"
1900 IF C=0 THEN LOCATE 5,8:PRINT"AGENDA
IS NOG LEEG !!!":FOR Y=1 TO 1000:NEXT Y
:GOTO 100
1910 LOCATE 5,8:PRINT"[1]....Wegschrijv
en naar TAPE"
1920 LOCATE 5,10:PRINT"[2]....Wegschrij
ven naar DISC"
1930 Q$=INKEY$:IF Q$="" THEN 1930
1940 IF Q$="1" THEN 1960
1950 IF Q$="2" THEN 1970 ELSE 1930
1960 LOCATE 5,10:PRINT Z$:GOTO 1980
1970 LOCATE 5,8:PRINT Z$
1980 IF Q$="1" THEN OPEN"cas:agenda" FOR
OUTPUT AS#1:GOTO 2000
1990 OPEN "agenda" FOR OUTPUT AS#1
2000 IF Q$="1" THEN MOTOR ON
2010 PRINT #1,C
2020 FOR Y=1 TO C
2030 PRINT #1,A$(Y)
2040 NEXT Y
2050 CLOSE #1
2060 IF Q$="1" THEN MOTOR OFF
2070 GOTO 100
2080 REM *****
      **          Agenda inlezen          **
      *****

2090 CLS:PRINT"BURO-AGENDA":PRINT STRING
$(39,"-"):PRINT:PRINT STRING$(39,"-")
2100 LOCATE 19,2:PRINT"AGENDA INLEZEN"
2110 LOCATE 5,8:PRINT"[1]....INLEZEN va
n TAPE"

```

```

2120 LOCATE 5,10:PRINT"[2].....INLEZEN v
an DISC"
2130 Q$=INKEY$:IF Q$="" THEN 2130
2140 IF Q$="1" THEN 2160
2150 IF Q$="2" THEN 2170 ELSE 2130
2160 LOCATE 5,10:PRINT Z$:GOTO 2180
2170 LOCATE 5,8:PRINT Z$
2180 IF Q$="1" THEN OPEN"cas:agenda" FOR
  INPUT AS#1:GOTO 2200
2190 OPEN "agenda" FOR INPUT AS#1
2200 IF Q$="1" THEN MOTOR ON
2210 INPUT #1,C
2220 FOR Y=1 TO C
2230 INPUT #1,A$(Y)
2240 NEXT Y
2250 CLOSE #1
2260 IF Q$="1" THEN MOTOR OFF
2270 GOTO 100
2280 REM *****
      ** Stoppen met Programma **
      *****
2290 CLS:PRINT"BURO-AGENDA":PRINT STRING
$(39,"-"):PRINT:PRINT STRING$(39,"-")
2300 LOCATE 19,2:PRINT"STOPPEN MET PROGR
AMMA"
2310 LOCATE 0,8:PRINT"U wilt stoppen met
  het programma ?":PRINT
2320 PRINT"Als U iets in de agenda veran
dert heeft is het verstandiger om deze a
genda eerstop te slaan op Tape of Disc !
!!":PRINT
2330 PRINT"Als dit niet nodig is of als
U dit al  gedaan heeft, Toetst U: [E]":
PRINT
2340 PRINT"In alle andere gevallen toets
t U [M]  voor terugkeer naar het Hoofd
menu !"
2350 LOCATE 0,22:PRINT" Toets Uw keuze i
n: [E] of [M]"

```



```
2360 Q$=INKEY$:IF Q$="" THEN 2360
2370 IF Q$="M" OR Q$="m" THEN 100
2380 IF Q$="E" OR Q$="e" THEN END ELSE 2
360
2390 GOTO 100
```

AANTEKENINGEN

AANTEKENINGEN

AANTEKENINGEN



WERKEN MET DE MSX COMPUTER; PRAKTIJKPROGRAMMA'S

De MSX computers kunnen zich verheugen in een steeds toenemende belangstelling, mede dankzij het feit dat er nu eindelijk een standaard voor homecomputers is ontwikkeld. Dit boek geeft de gebruiker kennis te maken met allerlei praktijkgerichte programma's. Een ideaal boek voor beginners die willen ontdekken hoe met gegevens moet worden gemanipuleerd om een bepaald resultaat te krijgen.

Daarom zijn er een aantal algemeen bruikbare programma's opgenomen zoals een Kaartenbak-programma, een Telefoonklapper met uitgebreide zoekmogelijkheden, een huis-houdboekje enz.

De programma's zijn allemaal geschreven in MSX-BASIC en geschikt voor alle merken MSX, o.a. SONY, PHILIPS, AVT, SPECTRAVIDEO, SANYO, TOSHIBA, GOLDSTAR, enz.

Terminal Software Publicaties is lid van de MSX Werkgroep Nederland.

TERMINAL SOFTWARE PUBLICATIES